$\mathcal{SageMath}^{\text{TM}}$ Advice

For Calculus

Tuan A. Le and Hieu D. Nguyen

Rowan University

Contents

Chapter 1

Introduction

1.1 SageMath

Welcome to SageMath! This tutorial manual is intended as a supplement to Rogawski's Calculus textbook and aimed at students looking to quickly learn Sage through examples. It also includes a brief summary of each calculus topic to emphasize important concepts. Students should refer to their textbook for a further explanation of each topic.

1.1.1 Creating an Account

SageMath is a powerful computer algebra system (CAS) whose capabilities and features can be overwhelming for new users. Thus, to make your experience in using Sage as easy as possible, we recommend that you read this introductory chapter carefully. We will discuss basic syntax and frequently used commands.

There are two ways to use Sage, you can run Sage on it server (cloud) or install Sage and run it on your computer:

SageMath Cloud: To use $\mathit{SageMath}$ on the cloud, go to **www.cloud.sagemath.com** and create an account. After logging in, you will see all of your projects will be listed. Since it's the first time, click on **NewProject**... to create one. Give the project a name and click on **CreateProject**. Your project now is created and listed under **ShowingProject**. For example, I have create a new project name "Testing Sage Manual" among other projects. The screen will look like this:



Click on the project you want to work on, click **Create or upload files...**



where we can create a file of upload a file from our computer. Since we want to run Sage on cloud, we create a new file name **StartingWithSage**, and select the type as **SageMath Worksheet**



We are now in a Sage file and ready to use it. Pay attention that we are now viewing the file **StartingWithSage.sagews** located inside of the **Testing Sage Manual** project.

<u>**Localhost:**</u> The other way to run Sage is to download it and install it on your computer. Go to **www.sagemath.org/download** and download Sage package. Install it and restart you computer. Now run Sage (double click on its icon), Sage will automatically open up your brownser



and your Sage notebook on your localhost, displayed all worksheets that you have been working on



Click on any worksheet that you want to continue work with or create a new worksheet. To create a new worksheet, click on **New Worksheet**, give it a name a click on **Rename**. For example, let create a new worksheet called **SageExample**



Each horizontal rectangle is called a cell. Click on that and you are now ready to start learning Sage.



1.1.2 Getting Started

Just start typing input commands (a cell formatted as an input box will be automatically created). For example, type $4+6$. To evaluate this command or any other command(s) contained inside an input box, simultaneously press SHIFT+ENTER, that is, the keys SHIFT and ENTER at the same time (or click on the **evaluate** button if you are on localhost or **run** button if you on the cloud). Be sure your mouse's cursor is positioned inside the input box or else select the input box(es) that you want to evaluate. This is how it looks like on localhost:



And on SageMath cloud:



Notice there is a slightly different between them.

1.1.3 Help Menu

SageMath provides an online help menu to answer many of your questions about the program. One can search for a particular command expression in the Help menu located at the right top conner.

For ony a brief description of **plot**, just evaluate **plot?**



1.1.4 Sharing Sage Files

SageMath Cloud not only lets you work anywhere as long as you have an Internet connection, but also allow you to share your file/project with your instructor or colleague. The only requirement is that the one who you want to share *Sage* files with should also have an account. Once he/she has it, you can give his/her permission to access your file. Notice that they have permission to access a particular file you choose, not every files you have in your account.

Once you sign in, click on the project (under **ShowingProject**) that you want to share, then click on **setting**. In **Collaborators** section, enter name or email address of your instructor or colleague, a list of matching will show up. Choose the one you look for and click on **Add selected**. That person will received an invitation email and now he/she can modify anything on that project. You and your instructor now can make a conversation or video call through the window of that project.

1.2 Sage Commands

1.2.1 Naming

Built-in *Sage* commands, functions, constants, and other expressions begin with lowercase letters and are (for the most part) one or more full-length English words (without capitalized). Furthermore, *Sage* is case sensitive. For example, **plot, expand, print** and **show** are valid function names. **sin, def, gcd** and **max** are some of the standard mathematical abbreviations that are exceptions to the full-length English word(s) rule.

User-defined functions and variables can be any mixture of uppercase and lowercase letter and number. However, a name cannot begin with a number. User-defined functions may begin with a upper case letter, but this is not requires. For example, **F1, g1, myPlot, Sol** and **Tech** are permissible function names.

1.2.2 Delimiters

Sage interprets various types of delimiters (brackets) differently.

- Parentheses, (): When there are multiple sets of parentheses in a formula, sometime mathematicians use brackets as a type of "strong parentheses". As it turns out, Sage needs the brackets for other things, like **list** or **table**, so you have to always use parentheses for grouping inside of formulas.

- Square brackets, []: It is used to construct a data structure with group of value such as a **list** or **table**.

1.2.3 Lists, Tables, and Arrays

Lists:

A list (or string) of elements can be defined in Sage as $[\mathbf{e_1}, \mathbf{e_2}, ..., \mathbf{e_n}]$. For example, the following command defines $v = [1,3,5,7,9]$ to be the list (set) of the first five odd positive integers.

```
sage: v=[1,3,5,7,9]
sage: v
[1, 3, 5, 7, 9]
```

To refer to the k^{th} element in a list name **expr**, just evaluate **expr[k]**. For example, to refer to the third element in v, we evaluate

```
sage: v[3]
7
```

It is also possible to define nested lists whose elements are themselves lists, call sublists. Each sublist contains subelements. For example, the list $w = [[1,3,5,7,9],[2,4,6,8,10]]$ contains two elements, each of which is a list (first five odd and even positive integers.)

```
sage: w=[[1,3,5,7,9],[2,4,6,8,10]]
sage: w
[[1, 3, 5, 7, 9], [2, 4, 6, 8, 10]]
```

To refer to the k^{th} subelement in the j^{th} sublist of **expr**, just evaluate **expr[j][k]**. For example, to refer to the fourth subelement in the second sublist of w (or 8), we evaluate

```
sage: w[1][3]
8
```

Tables:

A table is used to display a rectangular array or list as a table.

table(list)

For example, the following command displays v in a table.

```
sage: v=[['a','b','c,'],[1,2,3],[4,5,6]]
sage: table(v)
  a   b   c,
  1   2   3
  4   5   6
```

To highlight first row or first column, we set **header_row** = **True** or **header_column** = **True**, respectively. To put a box around each cell, set **frame** = **True**. Also, by default, **align** is 'left', we can change it to 'center' or 'right'. For example, let highlight the first row of the table of v, put a box around it, and align it center.

```
sage: table(v,header_row=True, frame=True, align='center')
+---+---+----+
| a | b | c, |
+===+===+====+
```

```
| 1 | 2 | 3  |
+---+---+----+
| 4 | 5 | 6  |
+---+---+----+
```

We can also use a loop inside table to create a table:

table([(x,f(x)) for x in [0..b]])

where b is number of counters or steps of x.

```
sage: table([(i,2*i) for i in [0..3]],frame=True)
+---+---+
| 0 | 0 |
+---+---+
| 1 | 2 |
+---+---+
| 2 | 4 |
+---+---+
| 3 | 6 |
+---+---+
```

Arrays:

Arrays are created using NumPy, that means you have to make numpy commands available in sage. You must first do: **import numpy**.

The following code will create an array called ArrayEx that contains the first 5 positive integers:

```
sage: import numpy
sage: ArrayEx=numpy.array([1,2,3,4,5])
sage: ArrayEx
[1 2 3 4 5]
```

To create a multiple array with the shape of 3x2 with the first column contains the first 3 integer and the second column contains double values of first column:

```
sage: import numpy
sage: ArrayMul=numpy.array([[j,2*j] for j in range(3)])
sage: ArrayMul
[[0 0]
 [1 2]
 [2 4]]
```

To refer to the k^{th} subelement in the j^{th} subarrays of **Array**, just evaluate **Array**[**j**][**k**]. For example, to refer to the second subelement in the third subarray of ArrayMul, we evaluate

```
sage: ArrayMul[2][1]
4
```

Notice that the index starts from 0.

1.2.4 Commenting

One can insert comments on any input line. The comments should be follow by # sign. For example,

```
sage: # This command plot the graph of $sin$ function in red
   color
sage: g=plot(sin(x),x,-3,3,figsize=3,color='red')
```



1.3 Algebra

1.3.1 Solving Equations

Sage uses the stand sysbols $+, -, *, /, \hat{}\,!$ for addition, subtraction, multiplication, division, raising powers (exponents), and factorials, respectively. Unlike other program, multiplication can only be performed by $*$ between factors.

To generate numerical output in decimal form, use the command $\mathtt{n}(\mathtt{expr}, \mathtt{digits} = 3)$ to display to 3 decimal places.

NOTE: Sage can perform calculations to arbitrary precision and handle numbers that are arbitrarily large or small.

```
sage: pi
pi
sage: n(pi,digits=4)
3.142
sage: n(pi,digits=20)
3.1415926535897932385
sage: 6^(5^2)
28430288029929701376
sage: factorial(5)
```

```
120
```

Here are *Sage* rules regarding the use of equal signs:

1) A single equal sign $(=)$ assigns a value to a variable. Thus, entering $x = 3$ means that x will be assigned the value 3.

```
sage: z=3
sage: z
3
```

If we then evaluate $5+z^3$, *Sage* will return 32

```
sage: 5+z^3
32
```

2) A double-equal sign $(==)$ is a test of equality between two expressions. Since we previously set $x = 2$, then evaluating $x == 2$ returns *True*, whereas evaluating $x == 3$ return *False*.

```
sage: x==2
x == 2
sage: x==3
x == 3
```

Another common usage of the double equal sign $(==)$ is to solve equations, such as the command **solve**$([x^2+x+1 == 0], x)$.

```
sage: solve([x^2+x+2==0],x)
[
x == -1/2*I*sqrt(7) - 1/2,
x == 1/2*I*sqrt(7) - 1/2
]
```

Sage is a host of built-in commands to help the user solve equations and manipulate expressions.

The command **solve(lhs==rhs, var)** solve the equation **lhs==rhs** for the variable **var**. For example, the command below solves the quadratic equation $x^2 - 2 = 0$ for x.

```
sage: solve(x^2-2==0,x)
[
x == -sqrt(2),
x == sqrt(2)
]
```

A system of m equations in n unknown can also be solved with using the same command, but formatted as

```
sage: x,y = var('x,y')
sage: solve([2*x-y==3,x+4*y==-2],x,y)
[
[x == (10/9), y == (-7/9)]
]
```

1.3.2 Useful Commands

In this section, we introduce few more popular commands in $Sage$.

- To simplify a function, we use **.simplify_full**() command :

f(x).simplify_full()

- To substitute a value c for variable x of a function, we use .**substitute**(**x** = **c**) command :

f(x).substitute(x=c)

- or substitute for multiple variable:

f(x,y).substitute(x=c,y=d)

- Define a function $f(x)$ such that $f(x) = f1$ on (a, b) and $f(x) = f2$ on (c, d), we use **Piecewise** command . Notice that unlike the other command in $Sage$, Piecewise command has the first letter capitalized:

f(x)=Piecewise([[(a,b),f1],[(c,d),f2]])

- To solve an equation $f(x) = 0$ for x, we use **solve** command:

solve(f(x)==0,x)

- To define y as a function of x:

y(x)=function('y',x)

- To factor a number or a function, we use **factor**() command :

factor(number)

- To expand an expression, we use **expand**():

expand(expression)

- To print a variable or a function f(x):

print f(x)

- To assign the right hand side of an equation contains in a variable u to x, we use **.rhs**() command:

x=u.rhs()

1.4 Functions

There are two ways to represent functions in Sage, depending on how they are to used. Consider the following example:

Example 1.4.1. Enter the function $\frac{x^2-x+4}{x-1}$ into Sage.

Solution:

Method 1: An explicitly way to present f as a function of the argument x is to enter:

```
sage: f(x)=(x^2-x+4)/(x-1)
sage: f(x)
(x^2 - x + 4)/(x - 1)
```

To evaluate $f(x)$ at $x = 5$, we use the command $\mathbf{f}(\mathbf{5})$

```
sage: f(5)
6
```

Method 2: Define a function as:

```
sage: def f(x): return (x^2-x+4)/(x-1)
sage: f(x)
(x^2 - x + 4)/(x - 1)
```

Example 1.4.2. Enter the following piece-wise function into Sage:

$$f(x) = \begin{cases} \tan(\pi x/4), \text{ it } |x| < 1 \\ x, \text{ if } |x| \geqslant 1 \end{cases}$$

Solution:

```
sage: def f(x):
....:     if abs(x) <1:
....:         return tan(π*x/4)
....:     else:
....:         return x
```

Chapter 2

Graphs, Limits, and Continuity of Functions

2.1 Plotting Graphs

2.1.1 Basics Plot

In this section, we will discuss how to plot graphs using Sage and how to utilize its various plot options. We will discuss in detail several options that will be useful in our study of calculus. The basic syntax for plotting the graph of a function $y = f(x)$ with x ranging in value from a to b is **plot(f,x,a,b)**.

plot(f(x),x,a,b)

Example 2.1.1. Plot the graph of $f(x) = 2x^2 - 4x + 2$ along the interval $[-3, 5]$

Solution:

```
sage: g=plot(2*x^2-4*x+2,x,-3,5)
```



Example 2.1.2. Plot the graph of $y = \sin(2x)$ along the interval $[-3, 3]$

Solution:

```
sage: g=plot(sin(2*x),x,-3,3)
```
91



Example 2.1.3. Plot the graphs of the two functions given in Example 1.1 and Example 1.2 prior on the same set of axes to show their points of intersection.

Solution:

```
sage: g=plot((2*x^2-4*x+2,sin(2*x)),x,-3,5)
```
92



Example 2.1.4. Plot the graphs of $f(x) = \frac{2x^2+x+2}{x+1}$ and $g(x) = \frac{\cos(2x)}{2}$ on the same set of axes.

Solution:

```
sage: g=plot(((2*x^2+x+2)/(x+1),(cos(2*x))/2),x,-5,5)
```
93



Note that the graph of $g(x) = \cos(2x)/2$ is displayed poorly in output above since its range (from -1 to 1) is too small compared to the range of $f(x) = (2x^2+x+2)/(x+1)$. We can zoom in by specify the value of vertical line using **ymin** and **ymax**.

```
sage: g=plot(((2*x^2+x+2)/(x+1),(cos(2*x))/2),x,-5,5,ymin=-10,
    ymax=6)
```
94



Example 2.1.5. Plot the graphs of the following functions.

(a) $f(x) = \frac{2x^2}{2-x^2}$ (b) $f(x) = 2\sin(x) + \cos(x)$ (c) $f(x) = xe^x + \ln x$ (d) $f(x) = \frac{2x^2}{x^2+2}$

Solution:

We recall that the natual base e is entered as e and that $\ln x$ is **log(x)**. Note that $\sin x$ and $\cos x$ are to be entered as $\sin(x)$ and $\cos(x)$.

(a)

```
sage: g=plot((2*x^2)/(2-x^2),x,-10,10,ymin=-4,ymax=4)
```



(b)

```
sage: g=plot(2*sin(x)+cos(x),x,-2*pi,2*pi)
```



(c)

```
sage: g=plot(x*e^x+log(x),x,-3,3,ymin=0,ymax=10)
```



(d)

```
sage: g=plot(2*x^2/(x^2+2),x,-4,4)
```



2.1.2 Plot Options

Next, we will introduce various options that can be specified within the **plot** command.

- Adding a **title** to a graph:

  plot(f(x),x,a,b,title="Here is a graph"

- Use **figsize** option to control the plot size:

  plot(f(x),x,a,b,figzie='a number'

- Draw a graph with **color**:

  plot(f(x),x,a,b, color= 'a color')

- Draw a graph and specify its **thickness**:

  plot(f(x),x,a,b,color= 'a color', thickness='a number')

- Draw graph with specify the **line style** and **legend_label**:

  plot(f(x),x,a,b,color= 'a color',linestyle='–', thickness='a number',legend_label='f(x)')

- Use **frame** option to puts a box around the graph

  plot(f(x),x,a,b,frame=True)

- Use **axes_labels** to verify the axes:

  plot(f(x),x,a,b,axes_labels=['x-axis, units','y-axis, units'])

- To draw an ellipse, use **implicit_plot** command:

implicit_plot(f(x),(x,a,b),(y,c,d))

Example 2.1.6. Plot$(x^3, x, -4, 4)$ with a title:"Here is a graph"

Solution:

```
sage: g=plot(x^3,x,-4,4, title="Here is a graph")
```
99



Example 2.1.7. Plot$(x^2, x, -4, 4)$ with different size.

Solution:

```
sage: g=plot(x^2,x,-4,4,figsize=2)
```
100




```
sage: g=plot(x^2,x,-4,4,figsize=4)
```
101



Example 2.1.8. Plot$(x^2, x, -4, 4)$ with purple color.

Solution:

```
sage: g=plot(x^2,x,-4,4, color= 'purple',figsize=3)
```



Example 2.1.9. Plot$(x^2, x, -4, 4)$ with color and thickness features.

Solution:

```
sage: g=plot(x^2,x,-4,4, color= 'green', thickness=10,figsize
    =3)
```



Example 2.1.10. Plot$(x^2, x, -4, 4)$ with multiple options.

Solution:

```
sage: g=plot(x^2,x,-4,4, color= 'green',linestyle='--',
   thickness=10, legend_label='f(x)',figsize=3)
```



Example 2.1.11. Plot multiple function on a single graphic:

Solution:

```
sage: g1=plot(x^3,x,-4,4, title='Here is a graph')
sage: g2=plot(x^2,x,-4,4, color= 'green',linestyle='--',
   thickness=2, legend_label='f(x)',figsize=3)
sage: g3=plot(2*x^2,x,-4,4, color= 'purple',figsize=3)
```

```
sage: g1+g2+g3
Graphics object consisting of 3 graphics primitives
```



Example 2.1.12. Plot(x^2, x, -4, 4) with color, frame, and label.

Solution:

```
sage: g=plot(x^2,x,-4,4, color= 'green',frame=True,
   legend_label='f(x)')
```



Example 2.1.13. Plot(x^2, x, -4, 4) with axes.

Solution:

```
sage: g=plot(x^2,x,-4,4,axes_labels=['x-axis,units','y-axis,
   units'])
```



Example 2.1.14. Draw the ellipse $\frac{x^2}{2} + \frac{y^2}{4} = 2$

Solution:

```
sage: x,y=var('x,y')
sage: g=implicit_plot(x^2/2+y^2/4==2, (x, -3, 3), (y, -3,3))
```



2.2 Limits

2.2.1 Evaluating Limits

To compute the limit of function $f(x)$ as x approaches a:

limit(f(x),x=a)

To compute the limit of function $f(x)$ as x approaches a from the left (meaning $x < a$):

limit(f(x),x=a,dir='minus')

To compute the limit of function $f(x)$ as x approaches a from the right (meaning $x > a$):

limit(f(x),x=a,dir='plus')

Example 2.2.1. Evaluate $\lim\limits_{x\to 1} \frac{2x^2+x+4}{x+1}$

Solution:

Following are tables of values of the function $\lim_{x\to 1} \frac{2x^2+x+4}{x+1}$ when x is sufficiently close to 1.

From the left:

```
sage: def f(x): return(2*x^2+x+4)/(x+1)
sage: step=float(1/100)
sage: initial=float(9/10)
sage: table([(i*step+initial,f(i*step+initial)) for i in
   [1..10]])
  0.91    3.43780104712
  0.92    3.44416666667
  0.93    3.45067357513
  0.94    3.45731958763
  0.95    3.4641025641
  0.96    3.47102040816
  0.97    3.47807106599
  0.98    3.48525252525
  0.99    3.49256281407
  1.0     3.5
```

From the right:

```
sage: def f(x): return(2*x^2+x+4)/(x+1)
sage: step=float(-1/100)
```

```
sage: initial=float(11/10)
sage: table([(i*step+initial,f(i*step+initial)) for i in
   [1..10]])
  1.09    3.57234449761
  1.08    3.56384615385
  1.07    3.5554589372
  1.06    3.54718446602
  1.05    3.53902439024
  1.04    3.53098039216
  1.03    3.52305418719
  1.02    3.51524752475
  1.01    3.50756218905
  1.0     3.5
```

From these tables, it is reasonable to expect that the limit is 3.5. Evaluating the limit confirm this:

```
sage: limit((2*x^2+x+4)/(x+1), x=1)
7/2
```

Example 2.2.2. Evaluate $\lim\limits_{x\to 1} \frac{2x^2+x-1}{x+1}$

Solution:

```
sage: limit((2*x^2+x-1)/(x+1),x=1)
1
```

Example 2.2.3. Evaluate $\lim\limits_{x\to 1^-} \frac{x^2-1}{x-1}$

Solution:

```
sage: limit((x^2-1)/(x-1),x=1, dir='minus')
2
```

Example 2.2.4. Evaluate $\lim_{x\to 1^+} \frac{x^2-1}{x-1}$

Solution:

```
sage: limit((x^2-1)/(x-1),x=1, dir='plus')
2
```

Example 2.2.5. Evaluate $\lim_{x\to -3} \frac{x+1}{x+3}$

Solution:

```
sage: limit((x+1)/(x+3),x=-3)
Infinity
```

Example 2.2.6. Show that $f(x) = 2*\cos(1/x)$ does not have a limiting value as x approach 0.

Solution:

We define

```
sage: f(x)=2*cos(1/x)
sage: initial=float(1/10)
sage: step=float(-1/100)
sage: table([(i*step+initial,f(i*step+initial)) for i in
    [1..8]])
  0.09   0.230559899091497
  0.08   1.99559655835716
  0.07   -0.296003263241933
  0.06   -1.14916333703824
```

```
0.05    0.816164123626784
0.04    1.98240562372695
0.03    -0.679423624807142
0.02    1.92993205698422
```

These values suggest that the limit does not exits. To make this clear, we consider the graph:

```
sage: g=plot(f(x),x,-1/10,1/10,figsize=3)
```



This indicate that there are too much oscillation around $x = 0$. Let us try to zooming in around this point:

```
sage: g=plot(f(x),x,-1/100,1/100,figsize=3)
```



Note that zooming in on this graph does not help. This indicates that the limit does not exist.

Example 2.2.7. Investigate the function $f(x) = \frac{x}{|x|}$ as $x \to 0$.

Solution:

```
sage: limit(x/abs(x),x=0,dir='minus')
-1
```

```
sage: limit(x/abs(x),x=0,dir='plus')
1
```

Since the left-hand and right-hand limits are not the same, we conclude that the limit does not exist.

```
sage: g=plot(x/abs(x),x,-3,3,figsize=3)
```



2.2.2 Limits Involving Trigonometric Functions

For trigonometric functions, Sage uses the same traditional notation in calculus.

Example 2.2.8. Evaluate $\lim_{x \to 0} \frac{\sin(3x)}{x}$

Solution:

```
sage: limit((sin(3*x)/x),x=0)
3
```

We can check the answer by graphing the function up close to the neighborhood of $x = 0$

```
sage: g=plot((sin(3*x)/x),x,-1,1)
```



Example 2.2.9. Evaluate $\lim_{t \to 0} \frac{\tan t}{|t|}$

Solution:

```
sage: limit((tan(x))/abs(x),x=0,dir='plus')
1
sage: limit((tan(x))/abs(x),x=0,dir='minus')
-1
```

Thus the limit does not exist. This can be clearly seen from the graph of the function below.

```
sage: g=plot((tan(x))/abs(x),x,-4,4,ymin=-10,ymax=10)
```



Example 2.2.10. Find

(a) $\lim_{x\to\pi} \frac{1}{\cos x}$ (b) $\lim_{x\to -\frac{\pi}{2}} 3^{\sin x}$

Solution:

```
sage: limit(1/cos(x), x=pi)
-1
sage: limit(3^(sin (x)),x=- pi/2)
1/3
```



Example 2.2.11. Find $\lim_{x\to c} \frac{\sin x - \sin c}{\sin c}$ for values of $c = 0, \pi/6, \pi/4\pi/3, \pi/2$.

Solution:

sage: c= $[0, \pi/6, \pi/4, \pi/3, \pi/2]$

sage: for i in range(5):

$\cdots$: limit((sin x - sin c[i])/(sin c[i]),x=c[i])

$0, -\frac{1}{2}, -\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{3}, -1$

Example 2.2.12. Find $\lim_{x\to 0} \frac{\cos(nx)-1}{x^2}$ for various values of n.

Solution:

Here is a table of limits for integer values of n ranging from 1 to 10. Notice that to avoid the confusing between an integer n and **n** command which returns numerical value, we always try to substitute integer n by i in sagecommandline:

```
sage: table([(limit((cos(i*x)-1)/x^2,x=0))for i in [1..10]])
  -1/2    -2    -9/2    -8    -25/2    -18    -49/2    -32    -81/2
     -50
```

A reasonable guess at a general formula for the answer would be $\lim_{x\to 0}(\cos(nx)-1)/x^2 = -n^2/2$. We can check this with values of n ranging from 10 to 20.

```
sage: table([([(limit((cos(i*x)-1)/x^2,x=0), -i^2/2)])for i in
    [10..20]])
  (-50, -50)
  (-121/2, -121/2)
  (-72, -72)
  (-169/2, -169/2)
  (-98, -98)
  (-225/2, -225/2)
  (-128, -128)
  (-289/2, -289/2)
  (-162, -162)
  (-361/2, -361/2)
  (-200, -200)
```

For a mathematical proof, first take $n = 1$ and plot the graph

```
sage: g=plot((cos(x)-1)/x^2,x,-pi,pi,figsize=3)
```



The graph above confirms that the limit is $-1/2$.

For the general case, let $t = nx$ so that $x^2 = \frac{t^2}{n^2}$. Then note that $x \to 0$ if and only if $t \to 0$. Thus, the limit can be evaluated in terms of t as

$$\lim_{x\to 0} \frac{\cos(nx)-1}{x^2} = \lim_{t\to 0} \frac{\cos(t)-1}{t^2/n^2} = n^2 \lim_{t\to 0} \frac{\cos(t)-1}{t^2} = -\frac{n^2}{2}$$

2.2.3 Limits Involving Infinity

Example 2.2.13. Evaluate $\lim_{x\to\infty} \frac{3x-2}{\sqrt{x^2+2}}$ and $\lim_{x\to-\infty} \frac{3x-2}{\sqrt{x^2+2}}$

Solution:

```
sage: limit((3*x-2)/sqrt(x^2+2), x=infinity)
3
```

```
sage: limit((3*x-2)/sqrt(x^2+2), x=-infinity)
-3
```

Observe how the two limits differ. The following graph confirms this.

```
sage: a=plot((3*x-2)/sqrt(x^2+2), x,-15,15,figsize=3)
```



Example 2.2.14. Evaluate $\lim_{x \to 3^-} \frac{\sqrt{9-x^2}}{x-3}$

Solution:

```
sage: limit(sqrt(9-x^2)/(x-3), x=3, dir='minus')
-Infinity
```

We plot the function over two different ranges to visually understand why the answer is -∞. Notice how the first range fails to show this.

```
sage: g1=plot(sqrt(9-x^2)/(x-3), x,1,3,ymin=-5,ymax=2,figsize
    =3)
```




```
sage: g2=plot(sqrt(9-x^2)/(x-3), x,1,3,ymin=-100,ymax=2,
    figsize=3)
```



Example 2.2.15. Evaluate $\lim_{x\to\infty} \cos(x)$

Solution:

```
sage: limit(cos(x), x=infinity)
ind
```

Example 2.2.16. Find $\lim_{x\to\infty} \frac{\cos x}{x}$

Solution:

```
sage: limit(cos(x)/x, x=infinity)
0
```

We can verify this limit by using the Squeeze Theorem. In this case, we take $f(x) = -\frac{1}{|x|}$, $g(x) = \frac{\cos x}{x}$, and $h(x) = \frac{1}{|x|}$. Then $f(x) \leqslant g(x) \leqslant h(x)$ since $-1 \leqslant \cos x \leqslant 1$

```
sage: g1=plot((-1/abs(x)),x,0,10,ymin=-1,ymax=1,color='green',
   figsize=3)
sage: g2=plot(cos(x)/x,x,0,10,ymin=-1,ymax=1,color='red',
   figsize=3)
sage: g3=plot(1/abs(x),x,0,10,ymin=-1,ymax=1,color='purple',
   figsize=3)
```

```
sage: g=g1+g2+g3
```



Since $-\frac{1}{|x|}$ and $\frac{1}{|x|}$ both approach 0 as $x \to \infty$, we conclude that $\frac{\cos x}{x}$ approaches zero as well.

Example 2.2.17. Evaluate $\lim_{x \to 1^+} (\frac{1}{\ln x} - \frac{1}{x-1})$

Solution:

```
sage: limit(1/log(x)-1/(x-1),x=1, dir='plus')
1/2
```

We can graph the function near $x = 1$ to visually understand why the answer is $1/2$:

```
sage: g=plot(1/log(x)-1/(x-1), x,0,1, figsize=3)
```



Note, however, that this example shows that $1/\ln x$ and $1/(x-1)$ both grow to ∞ at the same rate as $x \to 1^+$

Example 2.2.18. Let $f(x) = \frac{x^{2n}-1}{x^{2m}-1}$. Evaluate $\lim_{x\to 1} f(x)$ by substituting in various values of m and n.

Solution:

```
sage: table([[limit((x^i-1)/(x^j-1),x=1) for i in [1..8]] for
    j in [1..8]], align='center',frame=True, header_row=['i_1','
    i_2','i_3','i_4','i_5','i_6','i_7','i_8'], header_column=[''
    ,'j_1','j_2','j_3','j_4','j_5','j_6','j_7','j_8'])
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
|     || i_1 | i_2 | i_3 | i_4 | i_5 | i_6 | i_7 | i_8 |
+=====++=====+=====+=====+=====+=====+=====+=====+=====+
| j_1 ||  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_2 || 1/2 |  1  | 3/2 |  2  | 5/2 |  3  | 7/2 |  4  |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_3 || 1/3 | 2/3 |  1  | 4/3 | 5/3 |  2  | 7/3 | 8/3 |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_4 || 1/4 | 1/2 | 3/4 |  1  | 5/4 | 3/2 | 7/4 |  2  |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_5 || 1/5 | 2/5 | 3/5 | 4/5 |  1  | 6/5 | 7/5 | 8/5 |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_6 || 1/6 | 1/3 | 1/2 | 2/3 | 5/6 |  1  | 7/6 | 4/3 |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_7 || 1/7 | 2/7 | 3/7 | 4/7 | 5/7 | 6/7 |  1  | 8/7 |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
| j_8 || 1/8 | 1/4 | 3/8 | 1/2 | 5/8 | 3/4 | 7/8 |  1  |
+-----++-----+-----+-----+-----+-----+-----+-----+-----+
```

Can you guess a formula for $\lim_{x\to 1} f(x)$ in term of m and n? Enter the command **limit**$((\mathbf{x^n}-\mathbf{1})/(\mathbf{x^m}-\mathbf{1}),\mathbf{x}=\mathbf{1})$ into an input cell and evaluate it to verify your conjecture.

Let us end this section with an example where the **limit** command is used to evaluate the derivative of a function (in anticipation of commands introduced in the next chapter for computing derivaties). By definition, the derivative of a function f at x (i,e,m the slope of its tangent line at x) is

$$f'(x) = lim_{\Delta x\to 0}\frac{f(x+\Delta x)-f(x)}{\Delta x}$$

Example 2.2.19. Find the derivative of $f(x) = \frac{1}{4x}$ according to the limit definition.

Solution:

We first exam the derivative by tabulating values of the difference quotient, $\frac{f(x+\Delta x)-f(x)}{\Delta x}$, for some arbitrarily chosen values of Δx:

```
sage: f(x)= 1/(4*x)
sage: var('c')
c
sage: c=[10^(-1), 10^(-2), 10^(-4), 10^(-5), 10^(-6), 10^(-8)]
sage: table([(n(c[i],digits=4),((f(x+c[i])-f(x))/c[i])) for i
   in [0..5]])
  0.1000       25/(10*x + 1) - 5/2/x
  0.01000      2500/(100*x + 1) - 25/x
  0.0001000    25000000/(10000*x + 1) - 2500/x
  0.00001000   2500000000/(100000*x + 1) - 25000/x
  1.000e-6     250000000000/(1000000*x + 1) - 250000/x
  1.000e-8     2500000000000000/(100000000*x + 1) - 25000000/x
```

This table suggest that $f'(x) = -1/(4x^2)$ in the limit as $Deltax \to 0$. We confirm this with $Sage$:

```
sage: limit((f(x+Deltax)-f(x))/Deltax,Deltax=0)
```

```
-1/4/x^2
```

2.3 Continuity

Recall that a function is countinous at $x = a$ if and only if $\lim_{x\to a} f(x) = f(a)$. Graphically, this means that there is no break (or jump) in the graph of f at the point $(a, f(a))$. It is not possible to indicate this discontinuity using computer graphics for the situation whre the limit exists and the function is defined at a but the limit is not equal to $f(a)$. For other cases of discontinuity, computer graphics are very helpful.

To verify if a given function is continuous at a point, we evaluate its limit there and check if this limit is equal to the value of the function.

Example 2.3.1. Show that the function $f(x) = x^3 - 1$ is continuous everywhere.

Solution:

We could draw the graph and observe this fact. On the other hand, we can get Sage to check continuity:

```
sage: def f(x): return x^3-1
sage: var('c')
c
sage: bool(limit(f(x),x=c)==f(c))
True
```

This means that $\lim_{x\to c} f(x) = f(c)$ and hence f is continuous everywhere.

Example 2.3.2. Find point of discontinuity for each of the followin function:

(a) Let $f(x) = \begin{cases} \frac{x^2-1}{x-1}, \text{ if } x \neq 1 \\ 2, \text{ if } x = 1 \end{cases}$

(b) Le $g(x) = \begin{cases} \frac{x^2-1}{x-1}, \text{ if } x \neq 1 \\ 6, \text{ if } x = 1 \end{cases}$

Solution:

The piece-wise functions can be defined by using **if, else**:

(a) Define the function f:

```
sage: def f(x):
....:     if x<>1:
....:         return (x^2-1)/(x-1)
....:     else:
....:         return 2
```

Then we can check continuity of f at $x = 1$.

```
sage: bool(limit(f(x),x=1)==f(1))
True
```

Hence, the function is continuous at $x = 1$. For continuity at other points, we observe that the rational function $\frac{x^2-1}{x-1}$ simplifies to $x+1$ in this case (factor the numerator!) and thus is continuous at any point except $x = 1$. Thus, f is continuous everywhere. We can also confirm this by examining the graph of f below.

```
sage: g=plot(f(x),x,-6,6,figsize=3)
```



(b) As in part a, we define the function and consider continuity of g at $x = 1$:

```
sage: def g(x):
....:         if x<>1:
....:             return (x^2-1)/(x-1)
....:         else:
....:             return 6
sage: bool(limit(g(x),x=1)==g(1))
False
```

Thus, g is NOT continuous at $x = 1$. For continuity at other point, we again observe that the rational function $\frac{x^2-1}{x-1} = x+1$ and thus is continuous for $x \neq 1$.

Caution: The plot of the graph of g given below indicates (incorrectly) that g is continuous everywhere! Care must be taken when examining $Sage$ plots to draw conclusion about continuity.

```
sage: g=plot(g(x),x,-6,6,figsize=3)
```
257



Example 2.3.3. Let $f(x) = \begin{cases} \cos(\frac{1}{x}), \text{ if } x \neq 0 \\ 0, \text{ if } x = 0 \end{cases}$. Prove that for any number k between -1 and 1 there exists a value for c such that $f(c) = k$.

Solution:

Note: observe that f is not continous at $x = 0$ so the converse of the Intermediate Value Theorem does not hold.

For $k = 0$, we choose $c = 0$ so that $f(c) = 0$. For any nonzero k between -1 and 1, define $y =$

$\cos^{-1} k$ (using the principal domain of the cos function) and let $c = 1/y$. Then $f(c) = \cos(1/c) = \cos y = k$. The graph of f following shows that there are in fact infinitely many choices for c.

```
sage: g=plot(cos(1/x),x,-pi,pi,figsize=3)
```
258



Chapter 3

Differentiation

3.1 The Derivative

In this section, we introduce few more popular commands in $Sage$.

- To calculate the derivative of a function, use **diff**() or .**derivative**() command:

diff(f(x)) or f(x).derivative()

- To differentiate $f(x,y)$ with respects to x:

diff(f(x,y),x)

- To compute the n derivative respect to x:

diff(f(x),x,n)

3.1.1 Slope of Tangent

The most fundamental concepts in calculus is the derivative. Its definition is given by

$$f'(a) = \lim_{h \to 0} \frac{f(h+a) - f(a)}{h}$$

where geometrically $f'(a)$ is the slope of the line tangent to the graph of $f(x)$ at $x = a$, provided that the limit exists. We can view this graphically in the illustration below, where the tangent line (shown in blue) is viewed as a limit of secant lines (one shown in red) as $h \to 0$.



Example 3.1.1. Calculate the derivative of $f(x) = \frac{x^4}{3}$ at $x = 1$ using the point-wise definition of a derivative.

Solution:

We first use the **table** command to tabulate slopes of secant lines passing through the points at $a = 1$ and $a + h = 1 + h$ by choosing arbitrarily small values for h (taken as reciprocal powers of 10)

```
sage: a,x,i=var('a,x,i')
sage: f(x)=x^4/3
sage: a=1
sage: table([(n(1/(10^i),digits=4), n((f(a+1/(10^i))-f(a))
   /(1/(10^i)),digits=4)) for i in [1..5]])
  0.1000        1.547
  0.01000       1.353
  0.001000      1.335
  0.0001000     1.334
  0.00001000    1.333
```

Note that our use of the **table** command, which displays a list as an array of rectangular cells. From the table output, we may conclude that $f'(1) = 4/3$. A more rigorous approach is to algebraically simplify the difference quotient, $\frac{f(a+h)-f(a)}{h}$

```
sage: ((f(a+h)-f(a))/h).simplify_full()
1/3*h^3 + 4/3*h^2 + 2*h + 4/3
```

It is now clear that $\frac{f(a+h)-f(a)}{h} \to \frac{4}{3}$ as $h \to 0$. This can be checked using *Sage* **limit** command:

```
sage: limit((f(a+h)-f(a))/h,h=0)
4/3
```

Below is a plot of the graph of $f(x)$ (in black) and its corresponding tangent line (in blue), which also confirms our answer:

```
sage: g1=plot(f(x),x,-3,3,ymin=-3,ymax=5,figsize=3,color='
    black')
sage: ff(x)=diff(f(x))
sage: g2=plot(ff(a)*(x-a)+f(a),x,-3,3,ymin=-3,ymax=5,figsize
    =3)
```



Recall that the tangent line of $f(x)$ at $x = a$ is given by:

$$y = f'(a)(x-a) + f(a)$$

3.1.2 Derivative as a Function

The derivative is best represented of as a slope function, one that gives the slope of the tangent line at any point on the graph of $f(x)$ where this slope exists:

$$f'(x) = \lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$$

Example 3.1.2. Compute the derivative of $\cos(x^2)$ and evaluate it at $x = \sqrt{\pi/4}$

Solution:

```
sage: f(x)=cos(x^2)
sage: diff(f(x)).substitute(x=sqrt(pi/4))
-1/2*sqrt(2)*sqrt(pi)
```

where **substitute()** command inserts values of variable in () into function $f'(x)$.

Note: Observe that the derivative of $\cos(x^2)$ is NOT $-\sin(x^2)$ but $-2x\sin(x^2)$. This is because $\cos(x^2)$ is a composite function. It's a rule for differentiating composite functions, known as the Chain Rules.

Example 3.1.3. Compute the derivative of $f(x) = \begin{cases} \frac{\cos x}{x} \text{ if } x \neq 0 \\ 0 \text{ if } x = 0 \end{cases}$

Solution:

To define functions described by two different formulas over separate domains, we employ $Sage$ **Piecewise** command.

```
sage: f1(x)=cos(x)/x
sage: f2(x)=0
sage: f = Piecewise([[(0,0),f2],[(-infinity,0),f1],[(0,
    infinity),f1]] )
sage: f.derivative()
```

```
Piecewise defined function with 3 parts, [[(0, 0), x |--> 0],
    [(-Infinity, 0), x |--> -sin(x)/x - cos(x)/x^2], [(0, +
    Infinity), x |--> -sin(x)/x - cos(x)/x^2]]
```

Note: It is clear for $x \neq 0$ that the derivative is $-\frac{\sin(x)}{x} - \frac{\cos(x)}{x^2}$ as a result of the Quotient Rule . Notice that the fact that $f(0) = 0$ does not mean that f is a constant.

A plot of the graph of $f(x)$ reveals that it is discontinuous at $x = 0$, and thus not differentiable there:



Example 3.1.4. Find the equation of the tangent line to the graph of $f(x) = \sqrt{2x+2}$ at $x = 2$.

Solution:

Remember that the tangent line to a function $f(x)$ at $x = a$ is $L(x) = f(a) + f'(a)(x - a)$. Hear, $a = 2$:

```
sage: f(x)=sqrt(2*x+2)
sage: L(x)=f(2)+diff(f(x)).substitute(x=2)*(x-2)
sage: L(x)
1/6*sqrt(6)*(x - 2) + sqrt(6)
```

To see that $L(x)$ is indeed the desired tangent line, we will plot f and L together:

```
sage: g=plot((f(x),L(x)),x,0,4,figsize=3)
```



Example 3.1.5. Find an equation of the line passing through the point $P(2,-3)$ and tangent to the graph of $f(x) = x^2 + 1$

Solution:

Let us refer to $Q(a, f(a))$ as the point of tangentcy for our desired tangent line. To determine Q, we compute the slope of our desired tangent line from two different perspectives:

Slope of line segment PQ:

```
sage: var('a')
a
sage: f(x)=x^2+1
sage: m=(f(a)-(-3))/(a-2)
sage: m
(a^2 + 4)/(a - 2)
```

Derivative of $f(x)$ at $x = a$:

```
sage: f(x)=x^2+1
sage: diff(f(x)).substitute(x=a)
2*a
```

Equating the two formulas for slope above and solving for a yields:

```
sage: solve(m==diff(f(x)).substitute(x=a),a)
```

```
[
a == -2*sqrt(2) + 2,
a == 2*sqrt(2) + 2
]
```

Since there are two valid solutions for a, we have in fact found two such tangent lines. Their equations are given by:

$$y_1 = f'(a)(x-a)+f(a) \text{ as } a \to 2-2\sqrt{2}$$

$$y_2 = f'(a)(x-a) = f(a) \text{ as } a \to 2+2\sqrt{2}$$

```
sage: y1(x)=(diff(f(x)).substitute(x=a)*(x-a)+f(x).substitute(
   x=a)).substitute(a=2*sqrt(2)+2).simplify_full()
sage: y1(x)
4*x*(sqrt(2) + 1) - 8*sqrt(2) - 11
sage: y2(x)=(diff(f(x)).substitute(x=a)*(x-a)+f(x).substitute(
   x=a)).substitute(a=-2*sqrt(2)+2).simplify_full()
sage: y2(x)
-4*x*(sqrt(2) - 1) + 8*sqrt(2) - 11
```

Plotting these tangent lines together with the graph of $f(x)$ confirms that our solution is correct:

```
sage: g=plot((f(x),y1(x),y2(x)),x,-6,6,figsize=3)
```



3.2 Higher-Order Derivatives

Suppose we are interested in pursuing higher order derivatives of a function. The reasons are they relate to applications of minimum and maximum values, physical applications such as velocity and acceleration, or finding points of inflection.

Example 3.2.1. Compute the first eight derivatives of $f(x) = \cos(x)$. What is the 255^{th} derivative of f?

Solution:

Here are the first eight derivative of f:

```
sage: f(x)=cos(x)
sage: ([diff(f(x),x,i) for i in [1..8]])
[-sin(x), -cos(x), sin(x), cos(x), -sin(x), -cos(x), sin(x),
    cos(x)]
```

We observe from the output that the higher-order derivatives of f are periodic modulo 4, which means they repeat every four derivative. Since 255 has remainder 3 divided by 4, it follows that

$$f^{(255)}(x) = f^{(3)}(x) = \sin(x)$$

Of course, Sage can compute this derivative (see output below), but the pattern above gives us a more in-depth understanding of the higher-order derivatives of $\cos(x)$.

```
sage: diff(f(x),x,255)
sin(x)
```

Example 3.2.2. Compute the first three derivatives of $f(x) = x\sin(x)$

Solution:

We use the command **diff(f(x),x,n)** to compute the n^{th} derivative of f. Here, we set $n = 1, 2, 3$

```
sage: f(x)=x*sin(x)
sage: diff(f(x),x)
x*cos(x) + sin(x)

sage: diff(f(x),x,2)
-x*sin(x) + 2*cos(x)

sage: diff(f(x),x,3)
-x*cos(x) - 3*sin(x)
```

A quicker way to generate a list of higher-order derivatives is to use the **table** command. For example, here is a list of the first five derivatives of f:

```
sage: ([diff(f(x),x,i) for i in [1..5]])
[x*cos(x) + sin(x), -x*sin(x) + 2*cos(x), -x*cos(x) - 3*sin(x)
   , x*sin(x) - 4*cos(x), x*cos(x) + 5*sin(x)]
```

3.3 Chain Rule and Implicit Differentiation

In this section, we demonstrate not only how *Sage* uses the Chain Rule to differentiate composite functions but also to compute derivatives of functions defined implicitly by equations where solving for the dependent variable is not desirable.

Example 3.3.1. Find all horizontal tangents of $f(x) = \sqrt{\frac{2x^4-2x+1}{2x^4+x+1}}$

Solution:

We first compute the derivative of f, which requires the Chain Rule.

```
sage: f(x)=sqrt((2*x^4-2*x+1)/(2*x^4+x+1))
sage: diff(f(x)).simplify_full()
3/2*(6*x^4 - 1)/((4*x^8 + 4*x^5 + 4*x^4 + x^2 + 2*x + 1)*sqrt
    ((2*x^4 - 2*x + 1)/(2*x^4 + x + 1)))
```

Horizontal tangents have zero slope and so it suffices to solve $f'(x) = 0$ for x.

```
sage: solve(diff(f(x))==0,x)
[
x == 1/6*I*6^(3/4),
x == -1/6*6^(3/4),
x == -1/6*I*6^(3/4),
x == 1/6*6^(3/4)
]
```

Observe that the solutions above are nothing more than the zeros of the numerator of $f'(x)$. We ignore the first and third solutions listed above, which are imaginary. Hence, $x = \frac{1}{6} * 6^{\frac{3}{4}} = 0.6389$ and $x = -\frac{1}{6} * 6^{\frac{3}{4}} = -0.6389$. A plot of the graph of f below confirms our solution.

```
sage: g=plot(f(x),x,-2,2,figsize=3)
```



Example 3.3.2. Find all horizontal tangents of the lemniscate described by $4(x^2+y^2)^2 = 15(x^2-y^2)$

Solution:

Implicit differentiation is required here to compute $\frac{dy}{dx}$, which involves first differentiating the lemniscate equation and then solving for our derivative. Observe that we make the substitution $y \to y(x)$, which makes explicit our assumption that y depends on x.

```
sage: var('x,y')
(x, y)
sage: y(x)=function('y')(x)
sage: eq=4*(x^2+y^2)^2==15*(x^2-y^2)
sage: eq.substitute(y=y(x))
x |--> 4*(x^2 + y(x)^2)^2 == 15*x^2 - 15*y(x)^2
sage: diff(eq,x)
x |--> 16*(x^2 + y(x)^2)*(y(x)*D[0](y)(x) + x) == -30*y(x)*D
    [0](y)(x) + 30*x
sage: solve(diff(eq,x),diff(y(x)))
[
D[0](y)(x) == -(8*x^3 + 8*x*y(x)^2 - 15*x)/(8*y(x)^3 + (8*x^2
    + 15)*y(x))
]
```

Notice that D[0](y)(x) is the first derivative of $y(x)$ or $y'(x)$.

To find horizontals, it suffices to find where the numerator of $y'(x)$ vanishes (since the denominator never vanishes except when $y = 0$). Thus, we solve the system of equations

$$\begin{cases} 15x - 8x^3 - 8xy^2 = 0 \\ 4(x^2+y^2)^2 = 15(x^2-y^2) \end{cases}$$

since the solutions must also lie on the lemniscate.

```
sage: var('x,y')
(x, y)
sage: solve([4*(x^2+y^2)^2==15*(x^2-y^2),15*x-8*x^3-8*x*y^2==0],x,y)
[
[x == 0, y == -1/2*I*sqrt(15)],
[x == 0, y == 1/2*I*sqrt(15)],
[x == 0, y == 0],
[x == -3/8*sqrt(5)*sqrt(2), y == -1/8*sqrt(15)*sqrt(2)],
[x == -3/8*sqrt(5)*sqrt(2), y == 1/8*sqrt(15)*sqrt(2)],
[x == 3/8*sqrt(5)*sqrt(2), y == -1/8*sqrt(15)*sqrt(2)],
[x == 3/8*sqrt(5)*sqrt(2), y == 1/8*sqrt(15)*sqrt(2)]
]
```

From the output, we see that the last four solutions are valid:

$$(-3/8*\mathrm{sqrt}(5)*\mathrm{sqrt}(2), -1/8*\mathrm{sqrt}(15)*\mathrm{sqrt}(2)) \approx (-1.186, -0.685),$$

$$(-3/8*\mathrm{sqrt}(5)*\mathrm{sqrt}(2), 1/8*\mathrm{sqrt}(15)*\mathrm{sqrt}(2)),$$

$$(3/8*\mathrm{sqrt}(5)*\mathrm{sqrt}(2), -1/8*\mathrm{sqrt}(15)*\mathrm{sqrt}(2)),$$

$$(3/8*\mathrm{sqrt}(5)*\mathrm{sqrt}(2), 1/8*\mathrm{sqrt}(15)*\mathrm{sqrt}(2))$$

which can be confirmed by inspecting the graph of the lemniscate below. Observe the sysmmetry in the solutions.

```
sage: x,y=var('x,y')
sage: g=implicit_plot(4*(x^2+y^2)^2==15*(x^2-y^2), (x,-3,3),(y
    ,-2,2),figsize=3)
```



3.4 Derivatives of Inverse, Exponential and Logarithmic Functions

3.4.1 Inverse Function

Recall that a function $g(x)$ is the inverse of a given function $f(x)$ if $f(g(x)) = g(f(x)) = x$. The inverse of $f(x)$ is denoted by $f^{-1}(x)$. We note that a necessary and sufficient condition for a function to have an inverse is that it must be one-to-one. On the other hand, a function is one-to-one if it is strictly increasing or strictly decreasing throughout its domain.

Example 3.4.1. Determine if the function $f(x) = 2x^2 - 2x + 1$ has an inverse on the domain $(-\infty, \infty)$. If it exists, then find the inverse.

Solution:

We note that $f(0) = f(1) = 1$. Thus, f is not one-to-one. We can also plot the graph of f and note that it fails the Horizontal Line Test since it is not increasing on its domain.

```
sage: f(x)=2*x^2-2*x+1
sage: g=plot(f(x),x,-1,2,figsize=3)
```



However, observe that if we restrict the domain of f to an interval where f is either increasing or decreasing, say $[1,\infty]$, then its inverse exists:

```
sage: g=plot(f(x),x,1,5,figsize=3)
```



To find the inverse on this restricted domain, let $y = f^{-1}(x)$, Then $f(y) = x$. Thus, we solve for y from the equation $f(y) = x$.

```
sage: var('x,y')
(x, y)
```

```
sage: sol=solve(f(y)==x,y)
sage: sol
[
y == -1/2*sqrt(2*x - 1) + 1/2,
y == 1/2*sqrt(2*x - 1) + 1/2
]
```

Note that $Sage$ gives two solotions. Only the second one is valid because it has range $[1,\infty]$, which agrees with the domain of f. Therefore,

$$f^{-1}(x) = \frac{1}{2}(1+\sqrt{2x-1})$$

To extract this solution from the above output, we use the syntax below and denote the inverse function in $Sage$ by $g(x)$.

```
sage: g(x)=sol[1].rhs()
sage: g(x)
1/2*sqrt(2*x - 1) + 1/2
```

Note: One can also attempt to verify $g(f(x)) = x$. However, $Sage$ cannot confirm this identity:

```
sage: g(f(x)).simplify_full()
1/2*sqrt(4*x^2 - 4*x + 1) + 1/2
```

Lastly, a plot of the graph of $f(x)$ and $g(x)$ shows their expected symmetry about the diagonal line $y = x$.

```
sage: h=plot((f(x),g(x)),x,1/2,5,figsize=3,ymin=0,ymax=5)
```



Example 3.4.2. Determine if the function $f(x) = 2x^3 + 3x$ has an inverse. If it exists, then compute $(f^{-1})'(2)$.

Solution:

Since $f'(x) = 6x^2 + 3$, f is increasing on its domain therefore it has an inverse. Again, we can solve for this inverse as in the previous example:

```
sage: var('f,g,x,y,sol')
(f, g, x, y, sol)
sage: f(x)=2*x^3+3*x
sage: sol=solve(f(y)==x,y)
sage: sol
[
y == -1/2*(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)*(I*sqrt(3) + 1) +
   1/4*(-I*sqrt(3) + 1)/(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3),
y == -1/2*(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)*(-I*sqrt(3) + 1) +
    1/4*(I*sqrt(3) + 1)/(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3),
y == (1/4*x + 1/4*sqrt(x^2 + 2))^(1/3) - 1/2/(1/4*x + 1/4*sqrt
   (x^2 + 2))^(1/3)
]
```

Only the third solution listed above is valid, being real valued. Thus:

$$f^{-1}(x) = (\frac{x}{4} + \frac{\sqrt{x^2+2}}{4})^{\frac{1}{3}} - \frac{1}{2(\frac{x}{4} + \frac{\sqrt{x^2+2}}{4})^{\frac{1}{3}}}$$

Denote our inverse as:

```
sage: g(x)=sol[2].rhs()
sage: g(x)
(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3) - 1/2/(1/4*x + 1/4*sqrt(x^2
   + 2))^(1/3)
```

Lastly, we compute $g'(2)$:

```
sage: n(diff(g(x)).substitute(x=2),digits=3)
0.207
```

3.4.2 Exponential and Logarithmic Functions

One of the most important functions in mathematics and its applications is the exponential function. In particular, the natual exponential function $f(x) = e^x$, where

$$e = \lim_{x \to 0} (1+x)^{1/x} \approx 2.718$$

In $Sage$, we use the lower letter e to denote the Euler number:

```
sage: limit((1+x)^(1/x),x=0)
e
```

Every exponential function $f(x) = a^x$, $a \neq 1$, $a > 0$, has domain $(-\infty, \infty)$ and range $(0, \infty)$. It is also one-to-one on its domain. Hence, it has an inverse. The inverse of an exponential function $f(x) = a^x$ is called the logarithm function and its denoted by $g(x) = \log_a x$. The inverse of the natural exponential function is denoted by $g(x) = \ln x$ and is called the natural logarithm. In $Sage$, we use $\log(x)$ for $\ln x$. Below is a plot of the graphs of e^x and $\ln x$ in red and green, respectively.

Observe their symmetry about the dashed line $y = x$.

```
sage: x=var('x')
sage: h1=plot(e^x,x,-2,5,figsize=3,color='red',ymin=-2,ymax=5)
sage: h2=plot(log(x),x,-2,5,figsize=3,color='green',ymin=-2,
   ymax=5)
sage: h3=plot(x, x,-2,5,figsize=3,linestyle='--',color='black'
   ,ymin=-2,ymax=5)
```



Please refer to Section 3.9 of Rogawski's Calculus book for derivative formulas of general exponential and logarithmic functions.

Example 3.4.3. Compute derivative of the following functions.

(a) $f(x) = 2^x$ (b) $f(x) = 2x^2 + e^x$ (c) $f(x) = \ln x^3$

Solution:

We will input the functions directly and use the command **diff**. Note that $\log(x^3)$ should read as $\ln x^3$.

(a)

```
sage: diff(2^x)
2^x*log(2)
```

(b)

```
sage: diff(2*x^2+e^x)
4*x + e^x
```

(c)

```
sage: diff(log(x^3))
3/x
```

Example 3.4.4. Find point on the graph of $f(x) = x^2 e^{3x+5} + 3x$ where the tangent lines are parallel to the line $y = 3x - 1$.

Solution:

Since the slope of the given line equals 3 it suffices to solve $f'(x) = 3$ for x to locate these points(s).

```
sage: var('f,␣sol')
(f, sol)
sage: f(x)=x^2*e^(3*x+5)+3*x
sage: sol=solve(diff(f(x))==3,x)
sage: sol
[
x == (-2/3),
x == 0
]
```

Hence, there are two solotion: $(x1, f(x1))$ and $(x2, f(x2))$:

```
sage: var('x1,x2')
(x1, x2)
sage: x1=sol[0].rhs()
sage: x2=sol[1].rhs()
sage: f(x1)
4/9*e^3 - 2
```

```
sage: f(x2)
0
```

The plot below confirms that the two corresponding tangent lines (in green) are indeed parallel.

```
sage: y1= f(x1)+(diff(f(x)).substitute(x=x1))*(x-x1)
sage: y2= f(x2)+(diff(f(x)).substitute(x=x2))*(x-x2)
sage: g1=plot(y1,x,-1,1,color='green',figsize=3,ymin=-5,ymax
    =15)
sage: g2=plot(f(x),x,-1,1,figsize=3,ymin=-5,ymax=15)
sage: g3=plot(y2,x,-1,1,color='green',figsize=3,ymin=-5,ymax
    =15)
```



Chapter 4

Applications of the Derivative

We have seen how the derivative of a function is itself a function. This idea leads to many possible applications, some of which we will now explore with *Sage* to demonstrate its ability to manipulate and calculate complicated or tedious expressions.

4.1 Related Rates

Also notice that *Sage* will display the first derivative of function $S(t)$ as:

$$D[0](S)(t) = \mathtt{diff}(S(t))$$

Example 4.1.1. Let us assume a rubber ball is sitting out in the sun and that the heat causes its surface area the increase at the rate of 3 square centimeters per hour. How fast is the radius increasing when the radius is 2 centimeters?

To slove this problem, we will need the formula for the surface area of a sphere: $S = 4\pi r^2$. Here, the surface area S and the radius r are expressed as functions of t (time).

```
sage: var('t,S,r')
(t, S, r)
sage: r(t)=function('r',t)
```

```
sage: S(t)=function('S',t)
sage: sa=S(t)==4*pi*(r(t))^2
sage: dsa=diff(sa,t)
sage: dsa
D[0](S)(t) == 8*pi*r(t)*D[0](r)(t)
```

Now differentiate this formula and solve for $r'(t)$:

```
sage: sol=solve(dsa, diff(r(t)))
sage: sol
[
D[0](r)(t) == 1/8*D[0](S)(t)/(pi*r(t))
]
```

Since the output above is a nested list (each set of square braces denotes a list) and our solution, $\frac{S'(t)}{8\pi r(t)}$, represent the second element of the first list, we can extract it in order to define $r'(t)$ as follows:

```
sage: Dr(t)=sol[0].rhs()
sage: Dr(t)
1/8*D[0](S)(t)/(pi*r(t))
```

Note: we will use $Df(x)$ to denote the first derivative of function $f(x)$. As above, $Dr(t) = r'(t)$. Since we are given that $S'(t) = 3$ and $r(t) = 2$, we substitute these into the formula for $r'(t)$:

```
sage: n(Dr(t).substitute(r(t)==2, diff(S(t),t)==3),digits=3)
0.0597
```

Therefore, when the radius is 2, it is increasing at the rate of about 0.0597 cm per hour.

4.2 Extrema

We now consider how to find critical points and inflection points to determine extrema. Recall that critical points of a function are those for which $f'(x) = 0$ or for which $f'(x)$ does not exist. Similarly, inflection points occur where either $f''(x) = 0$ or where $f''(x)$ does not exist. Extrema occur at critical points, but not all critical points are extrema. An inflection point is a point $(c, f(c))$ where concavity changes; this occurs where $f''(c) = 0$ or where $f''(x)$ does not exist, and like critical points, not all points where $f''(x) = 0$ (or where $f''(x)$ does not exist) are inflection points.

Example 4.2.1. Find all local extrema and inflection points of $f(x) = 1/(x^2 + 1)$

Solution:

We first define $f(x)$ in *Sage*:

```
sage: var('x,f')
(x, f)
sage: f(x)=1/(2*x^2+1)
sage: g=plot(f(x),x,-2,2,figsize=3,ymin=0,ymax=1)
```



To find extrema of f, we locate its critical points, that is, those points where $f'(x) = 0$ or $f'(x)$ is undefined. We can solve the first case using *Sage*:

```
sage: Df(x)=diff(f(x),x)
sage: Df(x)
```

```
-4*x/(2*x^2 + 1)^2
sage: solve(Df(x)==0,x)
[
x == 0
]
```

Since $f'(x)$ is defined everywhere, it follows that there is exactly one critical point at $x = 0$, and at that point, there is a maximum, as can be seen from the graph above. We could also have used the second derivative test to confirm this:

```
sage: diff(f(x),x,2).substitute(x=0)
-4
```

Since the second derivative is negative at $x = 0$, the curve is concave down there. This means that we have a local maximum at $x = 0$.

To find the points of inflection, we locate zeros of the second derivative:

```
sage: solve(diff(f(x),x,2)==0,x)
[
x == -1/6*sqrt(6),
x == 1/6*sqrt(6)
]
```

To determine if these solutions are indeed inflection points, we need to check if there is a sign change in $f''(x)$ on either side of each.

```
sage: g=plot(diff(f(x),x,2),x,-2,2,figsize=3)
```



Notice from the graph above that $f''(x)$ changes from positive to negative at $x = -\frac{\sqrt{6}}{6}$ and from negative to positive at $x = \frac{\sqrt{6}}{6}$. Thus, both point $(-\frac{\sqrt{6}}{6}, f(-\frac{\sqrt{6}}{6}))$ and $(\frac{\sqrt{6}}{6}, f(\frac{\sqrt{6}}{6}))$ are inflection points.

4.3 Optimization

Extreme values of a function occur where the first derivative $f'(x) = 0$ or $f'(x)$ does not exist. This idea allows us to find maximum and minimum, a very important and widely applied in many applications. For example, in business, people want to maximize the profits and minimize the costs. In auto industry, we want to know what shape of the car will minimize the air resistant. There are many similar problems exist in many other fields. We will go over some of these applications in this chapter.

4.3.1 Traffic Flow

Example 4.3.1. Traffic flow along a major highway in Philly between 6 AM and 10 AM can be modeled by the function $f(t) = 20t - 40\sqrt{t} + 50$ (in miles per hour), where $t = 0$ corresponds to 6 AM. Determine when the minimum traffic flow occurs.

Solution:

Let us find plot the graph of $f(t)$

```
sage: var('f,t')
```

```
(f, t)
sage: f(t)= 20*t-40*sqrt(t)+50
sage: g=plot(f(t),t,0,4,figsize=3)
```



Note from the plot above that the average speed is decreasing between 6 AM to 7 AM and increasing after 7 AM.

At 6 AM the average speed is:

```
sage: f(0)
50
```

or 50 mph. At 7 AM the average speed is:

```
sage: f(1)
30
```

or 30 mph. To see how the average speed varies throughout the day we make a table of these values at each half hour from 6 AM to 10 AM:

```
sage: step=float(1/2)
sage: initial=float(0)
sage: table([(i*step+initial,n(f(i*step+initial),digits=4))
    for i in [0..8]],align='left')
   0.0    50.00
```

```
0.5   31.72
1.0   30.00
1.5   31.01
2.0   33.43
2.5   36.75
3.0   40.72
3.5   45.17
4.0   50.00
```

We can see from the table that the average speed quickly drops from 50 mph to 30 mph in the first hour and then gradually increases back up to 50 mph during the next 3 hours. If we want to verify that the minimum occurs at 7 AM (or $t = 1$), we can use calculus. Since extrema occur where the derivative is 0, we set the derivative equal to zero and solve for t:

```
sage: solve(diff(f(t),t)==0,t)
[
t == 1
]
```

Therefore the minimum does occur when $t = 1$ (at 7 AM) and from the table we see that the minimum average speed at this time is 30 mph.

4.3.2 Minimum Cost

Example 4.3.2. Imagine there is an island locate at $(0, 1500)$ and a mainline electronic connection point at $(5000, 0$ where the unit is in meter. What would be the cheapest way to connect the island and mainland if the cost to lay cable underwater is 36 and on land is 24? We can lay cable underwater from $(1500, 0)$ to $(x, 0)$ and then lay cable on land from $(x, 0)$ to $(5000, 0)$. The variable x can vary between 0 and 5000. What value of x would minimize the cost for laying this cable and what would that minimum cost be?

Solution:



First, we need to determine the cost. There are two parts: the underwater part and the overland part. The cost of underwater part called c1 is $36 times the distance d1 from $(0,1500)$ to $(x,0)$:

```
sage: var('x,c1')
(x, c1)
sage: c1(x)=36*sqrt(1500^2+x^2)
```

The overland cost called c2 is $24 times the distance d2 from $(x,0)$ to $(5000,0)$:

```
sage: var('x,c2')
(x, c2)
sage: c2(x)=24*(5000-x)
```

The total cost is:

```
sage: var('x,cost')
(x, cost)
sage: cost(x) = c1(x)+c2(x)
sage: cost(x)
-24*x + 36*sqrt(x^2 + 2250000) + 120000
```

We need to minimize this cost function. First, we graph it to see if it has a minimum:

```
sage: g=plot(cost(x),x,0,5000,ymin=150000,ymax=220000,figsize
   =3)
```



Notice that this cost function has its minimum somewhere between 1000 and 2000. Also, we will note that as x gets close to that minimum the tangent lines of $\mathtt{cost}(x)$ are getting close to horizontal. In other words, the minimum will occur at a point x for which the derivative is zero or horizontal. This is a calculus problem that we can solve.

Also notice that in this particular problem, **solve** command will not evaluate the solution. We have to use **find_root** to numerically approximate the solution:

```
sage: var('c')
c
sage: c=find_root(diff(cost(x)),0,10000)
sage: c
1341.6407865
sage: n(cost(c))
160249.223594996
```

The minimum occurs at $x = 1341.64$ meters and minimum cost is approximately $\$160,250$

4.3.3 Packaging (Minimum Surface Area)

Example 4.3.3. The cost of packaging in business is related to the surface area of the package. Minimizing the surface area will minimize the cost. Assuming that a Sumsung has a refrigerator product that needs to be packaged in a rectangular box having a square base. If the volume of the box is required to be 2 cubic meter, then find the dimensions of the box that will minimize its surface area.

Solution:

Let sides of the square base is x and the height of the box is y, then the volume of the box is given by x^2y and must equal 2 cubic meters.

```
sage: var('x,y,S')
(x, y, S)
sage: constraint=x^2*y==2
```

The surface area of the box is $S = 4xy + 2x^2$ and is the quantity that must be minimized, where the area of top and bottom sides are x^2 and the 4 sides each have area xy. Using our volume constraint, $x^2y = 2$, we can solve for y in terms of x:

$$y = \frac{2}{x^2}$$

```
sage: sol=solve(constraint,y)
sage: sol
[
y == 2/x^2
]
```

The surface area function can then be expressed as a function of x only:

$$S(x) = 4xy + 2x^2 = 4x(2/x^2) + 2x^2 = 8/x + 2x^2$$

```
sage: S(x)=(4*x*y+2*x^2).substitute(y==sol[0].rhs())
sage: S(x)
2*x^2 + 8/x
```

Again, we use the idea that extrema occur at points where the derivative is zero, we have:

```
sage: solve(diff(S(x),x)==0,x)
[
x == 1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3),
x == -1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3),
x == 2^(1/3)
]
```

This equation has 1 real and 2 imaginary solutions. We need only the real solution of $x = 2^{1/3}$. We compare with the plot to see the actual minimum:

```
sage: g=plot(S(x),x,0,10,ymin=0,ymax=100,figsize=3)
```



Alternatively, we could have used the second derivative test to show that a minimum occurs at $x = 2^{1/3}$:

```
sage: (diff(S(x),x,2)).substitute(x==2^(1/3))
12
```

Since $f''(2^{1/3}) > 0$, we know that the graph is concave up at $x = 2^{1/3}$ and hence must have a minimum there. Since $y = 2^{1/3}$ when $x = 2^{1/3}$, we conclude that the box with minimum surface area is a 2 cube meters with sides of $2^{1/3}$ meters.

4.3.4 Maximize Revenue

Example 4.3.4. Suppose a travel agency charges 500 per person for a charter flight if exactly 80 people sign up. However, if more than 80 people sign up, then the fare is reduced by 2 per person for each additional person over the initial 80. The travel agency wants to know how many people they should book to maximize revenue. Also, determine what that maximum revenue is and what the corresponding fare is for each person.

Solution:

Let x denotes the number of passenger above 80 and the revenue is the product of the number of people multiplied by the cost (fare) per person. If $R(x)$ is defined as the revenue function, then $R(x) = (80+x)(500-2x)$. We want to determine the maximum value of $R(x)$ for $x \geqslant 0$. Let consider the graph:

```
sage: var('x,R')
(x, R)
sage: R(x)=(80+x)*(500-2*x)
sage: g=plot(R(x),x,0,200,ymin=40000,ymax=60000,figsize=3)
```



From the plot above, we see that a maximum occurs at about 80 to 90. To confirm this, we first solve for the critical points:

```
sage: solve(diff(R(x),x)==0,x)
[
x == 85
]
```

Therefore the maximum does indeed occur at $x = 85$, and the maximum revenue is:

```
sage: R(85)
54450
```

or \$54450. Since $80 + x$ represents the number of customers, this occurs when 165 customers sign up for the flights. In this case, the cost per person is:

```
sage: (500-2*x).substitute(x==85)
330
```

or \$330 per person.

4.4 Newton's Method

4.4.1 Programing Newton's Method

Newton's Method is a technique for calculating zeros of a function based on the direction of its tangent lines (hence, it requires first derivative). It is a recursive routine. tedious to do by hand and easily to make mistake. However, it is simple to handle with $Sage$. We need initial guess value to start with or in other word, we need to guess where to solution's location is. This is because an initial approximation x_0 for that zero, say at $x = r$, is needed to start the recursion. For example, we can specify x_0 by examining the graph of the function to see where the zeros are approximately. Then the next approximation x_1 can be found by the recursive formula $x_1 = x_0 - f(x_0)/f'(x_0)$.

This process can be iterated using the general formula:

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)}$$

Under suitable conditions, the sequence of approximation $\{x_0, x_1, x_2, ..., \}$ (called Newton sequence) will converge to r. However the Newton Method does not guarantee the convergent, if the initial guess is not good (or not close enough to the zero) then it will diverges, meaning we will not able to find the solution.

Example 4.4.1. Approximate the zeros of the function $f(x) = \ln(6-x^2) - x$.

Solution:

```
sage: var('x,f')
(x, f)
sage: f(x)=log(6-x^2)-x
sage: g=plot(f(x),x,0,4,ymin=-5,ymax=5,figsize=3)
```



Clearly, there is one zero between 1 and 1.5 based on the graph above. To approximate this zero, we define a function **newtn** to perform the recursion:

```
sage: var('x,newtn')
(x, newtn)
```

```
sage: newtn(x)=x-f(x)/(diff(f(x)))
```
544

To generate the corresponding Newton sequence, we compute 8 iterates of this function starting with an initial guess of $x = 1.5$.

```
sage: xzero=float(15/10)
sage: for i in range(8):
....:         xzero=newtn(xzero)
....:         print xzero
  1.4009754666568441
  1.3977834736657635
  1.3977805354266575
  1.3977805354241768
  1.397780535424177
  1.3977805354241768
  1.397780535424177
  1.3977805354241768
```

Hence, if we stop at 6 decimal spaces then the zero of $f(x) = \ln(6 - x^2) - x$ is 1.397780.

4.4.2 Divergence

As mention earlier, Newton's Method does not alway work. For instance, the function $y = x^{1/3}$ clearly has a root at $x = 0$:

```
sage: g=plot(x^(1/3),x,0,1,figsize=3,ymin=0,ymax=1)
```
545



Yet, Newton's Method fails for any guess $x \neq 0$:

sage: $f(x) = x^{1/3}$

sage: newtn(x)=x-f(x)/(diff(f(x)))

sage: xzero=float(5/10)

sage: for i in range(8):

....: xzero=newtn(xzero)

....: print xzero

−1.0

2.0

−4.0

8.0

−16.0

32.0

−64.0

128.0

4.4.3 Slow Convergence

Even when Newton's Method works, sometimes the Newton sequence converges very slowly to the zero. Consider the following function:

```
sage: var('x,f')
(x, f)
sage: f(x)=x^3-2*x-2
sage: g=plot(f(x),x,-3,3,ymin=-20,ymax=5,figsize=3)
```



Clearly, there is a root between 1.5 and 2. If we use the **newtn** function with out guess at $x = 1$, we get quick convergence to root:

```
sage: f(x) = x^3 - 2*x - 2
sage: newtn(x)=x-f(x)/(diff(f(x)))
sage: xzero=float(1)
sage: for i in range(8):
....:         xzero=newtn(xzero)
....:         print xzero
 4.0
 2.8260869565217392
 2.1467190137392356
 1.8423262771400926
 1.772847636439238
 1.7693013974364495
 1.7692923542973595
```

1.7692923542386314

But if we choose our initial guess near 0.7, the convergence is much slower. (It took 20 iterations to have the accuracy as the 8th iteration above).

Chapter 5

Integration

5.1 Antiderivatives (Indefinite Integral)

Integral(f(x),x) give the indefinite integral (or antiderivative) of f with respect to x. The command **integral** can evaluate all rational functions and a host of transcendental functions, including exponential, logarithmic, trigonometric, and inverse trigonometric functions.

To integrate a function $f(x, y)$ respects to x:

integral(f(x,y),x)

To integrate a $f(x)$ over $[a, b]$:

integral(f(x,y),x,a,b)

Example 5.1.1. Evaluate $\int (x^3 - 3x + 2)dx$

Solution:

```
sage: integral(x^3-3*x+2,x)
1/4*x^4 - 3/2*x^2 + 2*x
```

Example 5.1.2. Evaluate $\int x(x^3+2)^2 dx$

Solution:

```
sage: integral(x*(x^3+2)^2,x)
1/8*x^8 + 4/5*x^5 + 2*x^2
```

Example 5.1.3. Evaluate $\int \frac{2x}{\sqrt{x+1}} dx$

Solution:

```
sage: integral(2*x/(sqrt(x+1)),x).simplify_full()
4/3*sqrt(x + 1)*(x - 2)
```

Example 5.1.4. Evaluate $\int 2x^2 \sin(x^3) dx$

Solution:

```
sage: integral(2*x^2*sin(x^3),x)
-2/3*cos(x^3)
```

Note: $Sage$ can certainly integrate much more complicated functions, including those that may require using any of the integration techniques discussed in your calculus textbook. We will consider some of these in Section 5.4. Also note that $Sage$ does not explicitly include the constant of integration C in its answer. We should always assume that this is implicitly part of the answer.

5.2 Riemann Sums and the Definite Integral

Review of Riemann Sums: A partition of a closed interval $[a,b]$ is a set $P = \{x_0, x_1, ..., x_n\}$ of points of $[a,b]$ such that

$$a = x_0 < x_1 < x_2 < ... < x_n = b$$

Given a function f on a closed interval $[a,b]$ and a partition $P = \{x_0, x_1, ..., x_n\}$ of the interval $[a,b]$, recall that Riemann sum of f over $[a,b]$ relative to P is a sum of the form

$$\sum_{i=1}^{n} f(x_i^*)\Delta x_i,$$

where $\Delta x_i = x_i - x_{i-1}$ and x_i^* is an arbitrary point in the ith subinterval $[x_{i-1}, x_i]$. We assume that $\Delta x_i = \Delta x \, \frac{b-a}{n}$ for all i. A Riemann sum is therefore an approximation to the area of the region between the graph of f and the x-axis along the interval $[a,b]$. The exact area is given by the definite integral of f over $[a,b]$, which is defined to be the limit of its Riemann sums an $n \to \infty$ and is denoted by $\int_a^b f(x)dx$:

$$\int_a^b f(x)dx = \lim_{n\to\infty} \sum_{i=1}^{n} f(x_i^*)\Delta x.$$

This definite integral exists provided the limits exists. For a continuous function f, it can be shown that $\int_a^b f(x)dx$ exists.

5.2.1 Riemann Sum Using Left Endpoints

A Rieman sum of a function f relative to a partition P can be obtained by considering rectangles whose heights are based on the left endpoint of each subinterval of P. This is done by setting $x_i^* = x_i = a_i + (b-a)/n$ for $i = 1, ..., n-1$, so that the corresponding height of each rectangle is given by $f(x_i)$. Let `leftrs` denotes the formula for a Riemann sum using left endpoint, we have:

```
sage: a,b,nn,f,x,i,leftrs,xstar=var('a,b,nn,f,x,i,leftrs,xstar
   ')
sage: f(x)=x
sage: d=(b-a)/nn
sage: xstar(i)=a+(i-1)*d
sage: leftrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
```

where $f(x)$ is a function of x, nn is number of subinterval. (Since in $Sage$, n is a special function so we avoid to use the same letter by indicate the number of subinterval by nn). Notice that as $i=1$, $xstar=a$ implies the height is $f(a)$ which correspond the left endpoint of the first rectangle.

Example 5.2.1. Let $f(x)=x^2+1$ on $[0,2]$ and let $P=0,1/n,2/n,\ldots,(n-1)/n$ be a partition of $[0,2]$

(a) Approximate $\int_0^2 f(x)dx$ by computing the Riemann sum relative to P using the left endpoint method.

(b) Plot the graph of f and the rectangles corresponding to the Riemann sum in part (a).

(c) Find the limit of the Riemann sum obtained in part (a) by letting $n\to\infty$

Solution:

(a)

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,
   xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: xstar(i)=a+(i-1)*d
sage: leftrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: table([(i,n(leftrs(0,2,i),digits=4)) for i in range
   (10,110,10)], header_row=['n','Riemann␣Sum'],frame=True)
+-----+-------------+
| n   | Riemann Sum |
+=====+=============+
| 10  | 4.280       |
+-----+-------------+
| 20  | 4.470       |
+-----+-------------+
| 30  | 4.535       |
```

40	4.568
50	4.587
60	4.600
70	4.610
80	4.617
90	4.622
100	4.627

Thus $\int_0^2 (x^2+1)\,dx \approx 4.627$ for $n = 100$ (rectangles).

(b) Following plot represents a plot of the rectangles corresponding to the Riemann sum in part (a) using left endpoint and $n = 4$



(c) Evaluate `leftrs` in the limit as $n \to \infty$

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: xstar(i)=a+(i-1)*d
sage: leftrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: limit(leftrs(0,2,nn),nn=infinity)
14/3
```

Thus, $\int_0^2 (x^2+1)dx = 14/3$

5.2.2 Riemann Sum Using Right Endpoints

We can similarly define a Riemann sum of f relative to a partition P by considering rectangles whose height are based on the right endpoint of each subinterval P. Let $rightrs$ denotes the formula for a Rieman sum using right endpoint, we have:

```
sage: a,b,nn,f,x,i,rightrs,xstar=var('a,b,nn,f,x,i,rightrs,xstar')
sage: f(x)=x
sage: d=(b-a)/nn
sage: xstar(i)=a+i*d
sage: rightrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
```

Notice that as $i = 1$, $xstar = a + d$ implies the height is $f(a + d)$ which corresponds the right endpoint of the first rectangle.

Example 5.2.2. Redo example 5.2.1 with right endpoint method.

Solution:

(a)

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: d=(b-a)/nn
sage: xstar(i)=a+i*d
sage: rightrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: table([(i,n(rightrs(0,2,i),digits=4)) for i in range(10,110,10)], header_row=['n','Riemann Sum'],frame=True)
+-----+-------------+
| n   | Riemann Sum |
+=====+=============+
| 10  | 5.080       |
+-----+-------------+
| 20  | 4.870       |
+-----+-------------+
| 30  | 4.801       |
+-----+-------------+
| 40  | 4.768       |
+-----+-------------+
| 50  | 4.747       |
+-----+-------------+
| 60  | 4.734       |
+-----+-------------+
| 70  | 4.724       |
+-----+-------------+
| 80  | 4.717       |
```

```
+-----+-------------+
| 90  | 4.711       |
+-----+-------------+
| 100 | 4.707       |
+-----+-------------+
```

Thus $\int_0^2 (x^2+1)dx \approx 4.707$ for $n = 100$ (rectangles).

(b) The following is a plot of the rectangles corresponding to the Riemann sum in part (a) using the right endpoint $n =$



(c) Evaluate rightrs in the limit as $n \to \infty$

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,
   xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: xstar(i)=a+i*d
sage: rightrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: limit(rightrs(0,2,nn),nn=infinity)
14/3
```

5.2.3 Riemann Sum Using Midpoints

For midpoint method, the ith subinterval is given by $x_i^* = x_i = a+(i+1/2)(b-a)/n$. Let $midrs$ denotes the formula for a Riemann sum using midpoint, we have:

```
sage: a,b,nn,f,x,i=var('a,b,nn,f,x,i')
sage: f(x)=x
sage: d=(b-a)/nn
sage: xstar(i)=a+(i-1/2)*d
sage: midrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
```

notice that as $i = 1$, $xstar = a+(i-1/2)d$ implies the height is between $f(a)$ (left endpoint) and $f(a+d)$ (right endpoint).

Example 5.2.3. Redo the example 5.2.1 with midpoint method.

Solution:

(a)

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,
   xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: xstar(i)=a+(i-1/2)*d
sage: midrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: table([(i,n(midrs(0,2,i),digits=4)) for i in range
   (10,110,10)], header_row=['n','Riemann␣Sum'],frame=True)
+-----+-------------+
| n   | Riemann Sum |
+=====+=============+
```

```
| 10  | 4.660       |
+-----+-------------+
| 20  | 4.665       |
+-----+-------------+
| 30  | 4.666       |
+-----+-------------+
| 40  | 4.666       |
+-----+-------------+
| 50  | 4.666       |
+-----+-------------+
| 60  | 4.667       |
+-----+-------------+
| 70  | 4.667       |
+-----+-------------+
| 80  | 4.667       |
+-----+-------------+
| 90  | 4.667       |
+-----+-------------+
| 100 | 4.667       |
+-----+-------------+
```

Thus, $\int_0^2 (x^2+1)\,dx \approx 4.666$ for $n = 30$ (rectangles).

(b) The graph



(c) Evaluate midrs in the limit as $n \to \infty$

```
sage: a,b,nn,f,x,i,leftrs,xstar,d=var('a,b,nn,f,x,i,leftrs,
   xstar,d')
sage: d=(b-a)/nn
sage: f(x)=x^2+1
sage: d=(b-a)/nn
sage: xstar(i)=a+(i-1/2)*d
sage: midrs(a,b,nn)=sum(f(xstar(i))*d,i,1,nn)
sage: limit(midrs(0,2,nn),nn=infinity)
14/3
```

5.3 The Fundamental Theorem of Calculus

The most important and elegant achievement in calculus is the **Fundamental Theorem of Calculus (FTC)**, which demonstrate that integration and anti-differentiation are equivalent. It expressed in two part:

Part I: Let f(x) is continuous on $[a, b]$, we have:

$$\int_a^b f(x)dx = F(b) - F(a)$$

where $F(x)$ is any antiderivative of $f(x)$.

Part II:

$$\text{If } F(x) = \int_a^x f(t)dt \text{ , then } F'(x) = f(x)$$

Example 5.3.1. Evaluate $\int_1^5 \frac{2x}{\sqrt{4x-1}}dx$

Solution:

```
sage: integral((2*x)/(sqrt(4*x-1)),x,1,2)
5/6*sqrt(7) - 1/2*sqrt(3)
```

Example 5.3.2. Evaluate $\int_{\sqrt{3}}^2 \frac{\sqrt{3x^2-2}}{2x}dx$

Solution:

```
sage: integral((sqrt(x^2-3))/(2*x),x,sqrt(3),2)
-1/12*sqrt(3)*pi + 1/2
```

Example 5.3.3. Approximate $\int_0^1 \cot x^2 dx$

Solution:

Here is an example of an integral that $Sage$ cannot evaluate exactly but return unevaluated integral.

```
sage: integral(tan(x^2),x,0,1)
integrate(tan(x^2), x, 0, 1)
```

However, a numerical approximation is still possible by using **n()** command:

```
sage: n(integral(tan(x^2),x,0,1))
0.398414444597
```

Example 5.3.4. Use the fact that if $m \leqslant f(x) \leqslant M \ \ \forall x \in [a,b]$, then $m(b-a) \leqslant \int\limits_a^b f(x)dx \leqslant M(b-a)$ to approximate $\int\limits_0^2 \sqrt{2x^3+1}dx$.

Solution:

We see that the function $f(x) = \sqrt{2x^3+1}$ is increasing on $[0,2]$. We can simply find $f'(x)$ and observe that $f'(x) > 0$ for all x.

```
sage: g=plot(sqrt(2*x^3+1),x,0,2,figsize=3)
```



Thus, $1 = f(0) \leqslant f(x) \leqslant f(2) = \sqrt{17}$ and therefore:

$$1(2-0) \leqslant \int\limits_0^2 \sqrt{2x^3+1}dx \leqslant \sqrt{17}(2-0)$$

$$2 \leqslant \int\limits_0^2 \sqrt{2x^3+1}dx \leqslant 2\sqrt{17}$$

Let $Sage$ confirms this:

```
sage: integral(sqrt(2*x^3+1),x,0,2)
integrate(sqrt(2*x^3 + 1), x, 0, 2)
```

Since $Sage$ dit not exactly evaluate it, we use the numerical approximation command **n**()

```
sage: n(integral(sqrt(2*x^3+1),x,0,2))
```

```
4.03659298666
```

Example 5.3.5. Let $f(x) = \sin(x^2)$ on $[0,2]$ and define $F(x) = \int_0^x f(t)dt = \int_0^x \sin(t^2)dt$.

(a) Plot the graph of f.

(b) Find the value's of x for which $F(x)$ starts to decrease.

Solution:

(a) Let plot the graph of f.

```
sage: var('f,t,g')
(f, t, g)
sage: f(x)= sin(x^2)
sage: g=plot(f(x),x,0,2,figsize=3)
```



(b) We can see that the graph of f is above the x-axis (positive area) for x between 0 and $\pi/2$, and below the x-axis for x between $\pi/2$ to 2. Thus, F begins to decrease at $x = \pi/2$.

5.4 Integration Techniques

In the text book, you will learn different technique to evaluate an integral. In $Sage$, we do not need to specify the technique. $Sage$ will automatically chooses an appropriate technique for the problem. However, if the integrals which will not be able to evaluated in term of elementary, $Sage$ will return the integral unevaluated.

Below, you will see some examples of integral that involves trigonometric functions, exponential, and logarithmic functions. If you wish to solve them by hand, some of them will require integration by part, partial fraction decompositions, or trigonometric substitutions.

Example 5.4.1. Evaluate $\int \frac{x^3}{(x^4+2)^2}\,dx$

Solution:

If do it by hand, this integral involves using the substitution method.. Let $u = x^4+2$, hence $du = 4x^3 dx$:

$$\int \frac{x^3}{(x^4+2)^2}dx = \frac{1}{4}\int \frac{du}{u^2} = \frac{1}{4}\int u^{-2}du = \frac{1}{4}\frac{u^{-2+1}}{(-2+1)} = \frac{1}{4}\frac{u^{-1}}{-1} = -\frac{1}{4u} = -\frac{1}{4(x^4+2)}$$

And by $Sage$ command:

```
sage: integral(x^3/(x^4+2)^2)
-1/4/(x^4 + 2)
```

Example 5.4.2. Evaluate $\int \frac{2x^5+x^2+x+1}{x^2-1}\,dx$

Solution:

This integral requires long division and partial fraction decomposition to be solved by hand. Apply long division, we have:

$$\frac{2x^5+x^2+x+1}{x^2-1} = 2x^3+2x+1+\frac{3x+2}{x^2-1} = 2x^3+2x+1+\frac{3x}{x^2-1}+\frac{2}{x^2-1}$$

Hence:

$$\int \frac{2x^5+x^2+x+1}{x^2-1}dx = \int \left[2x^3+2x+1+\frac{3x}{x^2-1}+\frac{2}{x^2-1}\right]dx = \int (2x^3+2x+1)dx + \int \frac{3x}{x^2-1}dx + \int \frac{2}{x^2-1}dx$$

$$\int (2x^3+2x+1)\,dx = \frac{1}{2}x^4+x^2+x$$
$$\int \frac{3x}{x^2-1}\,dx = \frac{3}{2}\int\frac{du}{u} = \frac{3}{2}\log(x^2-1) = \frac{3}{2}\log(x-1)+\frac{3}{2}\log(x+1)$$
$$\int \frac{2}{x^2-1}\,dx = \int\frac{2}{(x-1)(x+1)}\,dx = \int\left[\frac{1}{x-1}-\frac{1}{x+1}\right]dx = \int\frac{1}{x-1}\,dx - \int\frac{1}{x+1}\,dx = \log(x-1)-\log(x+1)$$

Therefore

$$\begin{aligned}\int \frac{2x^5+x^2+x+1}{x^2-1}\,dx &= \left(\frac{1}{2}x^4+x^2+x\right)+\left[\frac{3}{2}\log(x-1)+\frac{3}{2}\log(x+1)\right]+[\log(x-1)-\log(x+1)]\\ &= \frac{1}{2}x^4+x^2+x+\frac{5}{2}\log(x-1)+\frac{1}{2}\log(x+1)\end{aligned}$$

And by $Sage$ command:

```
sage: integral((2*x^5+x^2+x+1)/(x^2-1))                    702
1/2*x^4 + x^2 + x + 1/2*log(x + 1) + 5/2*log(x - 1)       703
```

Example 5.4.3. Evaluate $\int \frac{x^4+2x^3+3x+1}{(x^2+1)^2}\,dx$

Solution:

This integral involves long division, partial fraction decomposition, and inverse trigonometric fuc-ntions.Apply long division, we have:

$$\frac{x^4+2x^3+3x+1}{(x^2+1)^2} = 1+\frac{2x^3-2x^2+3x}{x^4+2x^2+1} = 1+\frac{2x^3+2x}{x^4+2x^2+1}-\frac{2x^2}{(x^2+1)^2}+\frac{x}{(x^2+1)^2}$$

Hence:

$$\int\frac{x^4+2x^3+3x+1}{(x^2+1)^2}\,dx = \int\left(1+\frac{2x^3+2x}{x^4+2x^2+1}-\frac{2x^2}{(x^2+1)^2}+\frac{x}{(x^2+1)^2}\right)dx$$

For the first term of the right hand side:

$$\int 1\,dx = x \tag{1}$$

For the second term of the right hand side, let $u = x^4+2x^2+1 \Rightarrow du = (4x^3+4x)dx = 4(x^3+$

$x)dx$. Therefore:

$$\int\left(\frac{2x^3+2x}{x^4+2x^2+1}\right)dx=\frac{2}{4}\int\frac{du}{u}=\frac{1}{2}\log(u)=\frac{1}{2}\log(x^2+1)^2=\log(x^2+1) \tag{2}$$

For the third term, let $x=\tan\theta \Rightarrow x^2+1=\tan^2\theta+1=\sec^2\theta$ and $dx=\sec^2\theta d\theta$. Hence:

$$\int\left(-\frac{2x^2}{(x^2+1)^2}\right)dx=-2\int\frac{\tan^2\theta}{(\sec^2\theta)^2}\sec^2\theta d\theta=-2\int\frac{\tan^2\theta}{\sec^2\theta}d\theta=-2\int\sin^2\theta d\theta$$
$$=-2\int\frac{1-\cos 2\theta}{2}d\theta=-\theta+\frac{1}{2}\sin 2\theta$$



So:

$$\int\left(-\frac{2x^2}{(x^2+1)^2}\right)dx=-\theta+\frac{1}{2}\sin 2\theta=-\arctan x+\sin\theta\cos\theta=-\arctan x+\frac{x}{\sqrt{x^2+1}}\frac{1}{\sqrt{x^2+1}}$$
$$=-\arctan x+\frac{x}{x^2+1} \tag{3}$$

For the fourth term, let $v=x^2+1 \Rightarrow dv=2xdx$. So:

$$\int\frac{x}{(x^2+1)^2}dx=\frac{1}{2}\int\frac{dv}{v^2}=\frac{1}{2}\int v^{-2}dv=-\frac{1}{2v}=-\frac{1}{2(x^2+1)} \tag{4}$$

From (1), (2), (3), and (4):

$$\int \frac{x^4+2x^3+3x+1}{(x^2+1)^2}dx = x+\log(x^2+1)-\arctan x+\frac{x}{x^2+1}-\frac{1}{2(x^2+1)}$$
$$= x+\log(x^2+1)-\arctan x+\frac{2x-1}{2(x^2+1)}$$

Or by $Sage$ command:

```
sage: integral((x^4+2*x^3+3*x+1)/(x^2+1)^2)
x + 1/2*(2*x - 1)/(x^2 + 1) - arctan(x) + log(x^2 + 1)
```

Example 5.4.4. Evaluate $\int 2x^2\cos(x)dx$

Solution:

This integral requires integration by part technique. Let $u = x^2 \Rightarrow du = 2xdx$ and $dv = \cos(x)dx \Rightarrow v = \sin(x)$. Hence

$$\int 2x^2\cos(x)dx = 2\left(x^2\sin(x)-\int \sin(x)2xdx\right)$$

We again apply the integral by part method on $\int 2x\sin(x)dx$. Let $u_1 = x \Rightarrow du_1 = dx$ and $dv_1 = \sin(x)dx \Rightarrow v_1 = -\cos(x)$. Therefore

$$\int 2x^2\cos(x)dx = 2\left(x^2\sin(x)-\int \sin(x)2xdx\right) = 2\left[x^2\sin(x)-2\left(-x\cos(x)+\int\cos(x)dx\right)\right]$$
$$= 2x^2\sin(x)+4x\cos(x)-4\sin(x) = 4x\cos(x)+2\sin(x)(x^2-2)$$

And by $Sage$ command:

```
sage: integral(2*x^2*cos(x))
4*x*cos(x) + 2*(x^2 - 2)*sin(x)
```

Example 5.4.5. Evaluate $\int \frac{-4}{\sqrt{1-x^2}}dx$

Solution:

This integral involves trigonometric substitution.Let $x = \sin\theta \Rightarrow dx = \cos\theta d\theta$. Hence

$$\int \frac{-4}{\sqrt{1-x^2}}dx = -4\int \frac{dx}{\sqrt{1-x^2}} = -4\int \frac{\cos\theta d\theta}{\sqrt{\cos^2\theta}} = -4\int d\theta = -4\theta = -4\arcsin x$$

By *Sage* command:

```
sage: integral(-4/(sqrt(1-x^2)))
-4*arcsin(x)
```

Following are some examples of integrals that are important in applications but do not have an elementary antiderivative. The integral does not have closed-form expression ,i.e., the antiderivative can not be expressed in term of elementary functions (such as polynomial, logarithm, exponential, trig functions). For instance, this integral contain an error function **erf**- a special non-elementary function:

```
sage: integral(sin(x^2))
1/16*sqrt(pi)*((I + 1)*sqrt(2)*erf((1/2*I + 1/2)*sqrt(2)*x) +
   (I - 1)*sqrt(2)*erf((1/2*I - 1/2)*sqrt(2)*x) - (I - 1)*sqrt
   (2)*erf(sqrt(-I)*x) + (I + 1)*sqrt(2)*erf((-1)^(1/4)*x))
```

Notice how *Sage* returns the answer in terms of imagination numbers.

```
sage: integral(e^(-x^2))
1/2*sqrt(pi)*erf(x)
```

Where **erf** is an error function. It plays an important role in physics and engineering.

```
sage: integral(sin(x)/x)
-1/2*I*Ei(I*x) + 1/2*I*Ei(-I*x)
```

However, we can use **n()** to evaluate these integrals over any finite interval. For example:

```
sage: n(integral(e^(-x^2),x,0,10))
```

```
0.886226925452758

sage: n(integral(log(x)/x,x,2,50))
7.41173549054043
```

Chapter 6

Applications of the Integral

6.1 Area Between Curves

First, let us consider the problem of finding the area between two curves.

Example 6.1.1. Determine the area of the region bounded between the curves $f(x) = \frac{1}{2}\sin(x)$ and $g(x) = \csc^2(x)$ on $[\pi/4, \pi/2]$

Solution:

We first plot graphs of f and g.

```
sage: f(x)= 1/2*sin(x)
sage: g(x)=csc(x)^2
sage: h=plot((f(x),g(x)),x,pi/4,3*pi/4,figsize=3,fill=True)
```



Recall that $\csc(x)$ is greater than 1 in this interval. Hence, $\csc^2(x)$ is greater than $\sin(x)$ since $-1 \leqslant \sin(x) \leqslant 1$. Therefore, to calculate the area between $f(x)$ and $g(x)$ on this interval is:

```
sage: integral(g(x)-f(x),x,pi/4,3*pi/4)
-1/2*sqrt(2) + 2
```

Example 6.1.2. Determine the area of the region enclosed between the curves $f(x) = 2x(x^2 - 4x + 2)$ and $g(x) = x^2$

Solution:

```
sage: f(x)=2*x*(x^2-4*x+2)
sage: g(x)=x^2
sage: h=plot((f(x),g(x)),x,-2,6,figsize=3,ymin=-10,ymax=20)
```



The bounded region between the two curves seems to be at 0, 1/2 and 4. To make sure this, we solve for the intersection points:

```
sage: solve(f(x)==g(x),x)
[
x == 4,
x == (1/2),
x == 0
```

```
]
```

Hence, the intersection points are at $x = 0, 1/2, 4$. Notice that $f(x)$ is greater than $g(x)$ on $[0, 1/2]$ and $g(x)$ is greater than $f(x)$ on $[1/2, 4]$. Therefore the area enclosed between those curves is:

```
sage: integral(f(x)-g(x),x,0,1/2) + integral(g(x)-f(x),x
    ,1/2,4)
517/16
```

Example 6.1.3. Determine the area of the region bounded between the curves $f(x) = |2x|$ and $g(x) = \sin(x)$ on $[-\pi/2, \pi/2]$

Solution:

First we plot the graph:

```
sage: f(x)= abs(2*x)
sage: g(x)=cos(x)
sage: h=plot((f(x),g(x)),x,-pi/2,pi/2,figsize=3,fill=True)
```



From the graph, we will need to consider three separate areas. Note that the command **solve** does not work here because it is only able to solve algebraic equations. Instead, we use the **find_root** command to solve the equation $f(x) - g(x) = 0$, providing the interval where the root could be found.

```
sage: find_root(f(x)-g(x),0,1)
```

```
0.450183611295
```

Thus the approximately root is $a = 0.45018$. By symmetry, we have another root at $a = -0.45018$. Therefore, the area between these two functions is the sum of three integrals:

```
sage: a=float(45/100)
sage: n(integral(f(x)-g(x),x,-pi/2,a)+integral(g(x)-f(x),x,-a,
   a)+integral(f(x)-g(x),x,a,pi/2))
3.39973326876714
```

The area of the bounded region is 3.3997.

6.2 Average Value

Recall that the average value of a function $f(x)$ on $[a,b]$ is defined as:

$$f_{ave} = \frac{1}{b-a}\int_a^b f(x)dx$$

Also, remember that The Mean Value Theorem for Integrals state that for any continuous functions on $[a,b]$ there exists a value $c \in [a,b]$ such that:

$$f(c) = f_{ave}$$

Example 6.2.1. Let $f(x) = 3\sin(x) - x$

(a) Find the only positive root α of f.

(b) Calculate the average value of f on $[0,\alpha]$.

(c) Determine a value c that satisfies the Mean Value Theorem for Integral on $[0,\alpha]$.

Solution:

(a) Draw the graph:

```
sage: f(x)=3*sin(x)-x
sage: h=plot(f(x),x,-3,3,figsize=3)
```



Then use **find_root** command with the interval $[2,3]$ as our initial guess:

```
sage: find_root(f(x),2,3)
2.27886266008
```

Therefore $\alpha = 2.27886$ accurate to 5 decimal places.

(b) We calculate the average value of f on $[0,\alpha]$:

```
sage: alpha=float(227886*(10^(-5)))
sage: fave=1/(alpha-0)*integral(f(x),x,0,alpha)
sage: fave
1.033188037358966
```

Thus, the average value is approximately $f_{ave} = 1.033188$.

(c) By Mean Value Theorem of Integrals, there exists a value $c \in [0,\alpha]$ such that $f(c) = f_{ave}$. We can solve for c by this equation:

```
sage: var('c,x')
(c, x)
sage: find_root(f(c)-fave,0,1)
0.559759684314
```

6.3 Volume of Solids of Revolution

Recall the definition to evaluate the integral:

$$\int_a^b f(x)dx = \lim_{n\to+\infty}[\sum_{i=1}^{n} f(x_i^*)\Delta x_i]$$

Other important application of the definite integral involves finding the volume of a solid of revolution, that is, a solid obtained by revolving a region in the plane about one of the x or y axes.

6.3.1 The Methods of Discs

Suppose we have $y = f(x)$, $y = 0$, and two vertical lines $x = a$ and $x = b$. Let S be a solid of revolution obtained by revolving the region bounded by y about the x-axis. To obtain the volume of S, we can approximate S by discs, that is, the cylinder obtained by revolving each rectangle, constructed by a Riemann sum of f relative to a partition $P = (x_0, x_1, x_2, ..., x_n)$ of $[a, b]$, about the x-axis. Let the radius of the cylinder be R, the height is h, then the volume is:

$$V = \pi R^2 h$$

it means that the volume of the ith cylinder which corresponding to the ith rectangle is $V_i = \pi[f(x_i^*)]^2\Delta x$. So, an approximation to the volume of S is given by the Riemann sum:

$$Vol(S) \approx \sum_{i=1}^{n} V_i = \pi\sum_{i=1}^{n}[f(x_i^*)]^2\Delta x$$

As $n \to \infty$, we obtain the exact volume of S:

$$Vol(S) = \pi \lim_{n\to\infty}\sum_{i=1}^{n}[f(x_i^*)]^2\Delta x = \pi\int_a^b [f(x)]^2 dx$$

Notice that if the region is revolved about the y-axis then the volume of S is:

$$\mathrm{Vol}(S) = \pi \int_{c}^{d} [f(y)]^2 dy$$

Example 6.3.1. Find the volume of the solid of revolution obtained by rotating the region bounded by the graph of $f(x) = \sqrt{x+1}$, the x-axis, and the vertical line $x = 2$

Solution:

```
sage: var('u')
u
sage: f(u)=sqrt(u+1)
sage: h=plot(f(u),u,0,2,figsize=3,fill=True)
```



The plot show our region shaded in gray. Now, we rotate this shaded region about the x-axis to obtain a solid of revolution. In $Sage$, we use the **revolution_plot3d**$(\mathbf{f}(\mathbf{x}), \mathbf{x}, \mathbf{a}, \mathbf{b})$ command which generates a surface if revolution with radius **f** at height **x**

```
sage: s=revolution_plot3d(f(u),(u,0,2), show_curve=True,
   opacity=7, parallel_axis='x' ).show(aspect_ratio=(1,1,1))
```



```
sage: pi*integral(f(u)^2,u,0,3)
15/2*pi
```

6.3.2 The Method of Washers

If a solid of revolution S is generated by revolving a region bounded between two different curves $f(x)$ and $g(x)$ on $[a, b]$ about the x-axis, we use washer method. The corresponding volume of S is given by:

$$\mathrm{Vol}(S) = \pi \int_a^b [g(x)]^2 - [f(x)]^2 dx$$

given that $g(x) > f(x)$.

Example 6.3.2. Find the volume of the solid generated by revolving about the x-axis the region enclosed by $y = 2x^2 + 1$ and $y = x + 2$.

Solution:

```
sage: var('u')
```

```
u
sage: f(u)=2*u^2+1
sage: g(u)=u+2
sage: h=plot((f(u),g(u)),u,-2,2,figsize=3, ymin=0,ymax=5,fill=
    True)
```



We need the intersection points:

```
sage: solve(f(u)==g(u),u)
[
u == 1,
u == (-1/2)
]
```

We can easily verify that the intersection points are $(-1/2, 3/2)$ and $(1, 3)$. If we let S be the solid obtains by rotating the region between $f(x)$ and $g(x)$ on $[-1/2, 1]$ about the x-axis, then it can be view as the difference of the solid F obtains by rotating $f(x)$ and the solid G obtains by rotating $g(x)$ on that same interval:

```
sage: var('u,F')
(u, F)
sage: f(u)=2*u^2+1
```

```
sage: F=revolution_plot3d(f(u),(u,-1/2,1), show_curve=True,
   opacity=7, parallel_axis='x' )
```



```
sage: var('u,G')
(u, G)
sage: g(u)=(u+2)
sage: G=revolution_plot3d(f(u),(u,-1/2,1), show_curve=True,
   opacity=7, parallel_axis='x' )
```



```
sage: S=G+F
sage: S.show()
None
```



Since the curve $f(x)$ is lower than $g(x)$, the volume of S is given by:

```
sage: pi*integral((g(u)^2-f(u)^2),u,-1/2,1)
```

```
81/20*pi
```
785

6.3.3 The Method of Cylindrical Shells

Another approach to finding the volume of a solid of revolution is to approximate it using cylindrical shells. Recall that with dish method or washers method, we rotate the function on an interval around an axis. In cylindrical shells method, we rotate the rectangular of area whose height is paralell to the axis of rotation.

A cylindrical shell is defined as a solid generated by two cylinders having the same axis of rotation. Suppose a cylindrical shell has an inner redius r_1 and outer radius of r_2 with altitude h, then the volume is defined as:

$$\mathrm{Vol} = \pi\, r_2^2\, h - \pi\, r_1^2\, h = 2\,\pi\,\bar{r}\,h\,\Delta x$$

where $\bar{r} = \frac{r_1+r_2}{2}$: the average of radius and $\Delta x = r_2 - r_1$

Assume we have a function $f(x)$ defined on $x = a$ and $x = b$. Let S is the solid obtain by rotate the region between $f(x)$, x-axis, a and b about y-axis. Then the volume of ith shell is the corresponding ith rectangle and defined as:

$$\mathrm{Vol}_i = 2\,\pi\,x_i^*\;f(x_i^*)\;\Delta x$$

where $x_i^* = (x_i - x_{i-1})/2$. Therefore:

$$\mathrm{Vol}(S) \approx \sum_{i=1}^{n} \mathrm{Vol}_i = 2\,\pi\,\sum_{i=1}^{n}\;x_I^*\;f(x_I^*)\;\Delta x$$

As $n \to \infty$, we obtain the exact volume of S:

$$\mathrm{Vol}(S) = 2\,\pi\,\lim_{n\to\infty}\,\sum_{i=1}^{n} x_I^*\;f(x_I^*)\;\Delta x = 2\,\pi\int_a^b x\;f(x)\;dx$$

Similarly, if the region is rotated about the x-axis then the volume of S is given by:

$$\mathrm{Vol}(S) = 2\,\pi \int_{c}^{d} y\, f(y)\, dy$$

Example 6.3.3. Consider the region bounded by the curve $y = x^2 + 1$, the x-axis, and the line $x = 3$. Find the volume of the solid generated by revolving this region about the y-axis using the method of cylindrical shells.

Solution:

```
sage: var('u,f')
(u, f)
sage: f(u)=u^2+1
sage: h=plot(f(u),u,0,3,figsize=3,fill=True)
```



We then revolve this shaded region about the y-axis to obtain the solid S. Let Q be the cylinder when we rotate $x = 3$ and P the paraboloid of rotating $f(x)$ about y-axis, then S can be seen as Q with P removed from it:

```
sage: var('u,P')
(u, P)
sage: f(u)=u^2+1
```

```
sage: P=revolution_plot3d(f(u),(u,0,3), show_curve=True,
    opacity=7)
```



```
sage: var('u,Q,f')
(u, Q, f)
sage: Q=revolution_plot3d((3,f),(f,0,10), show_curve=True,
    opacity=7 )
```



```
sage: S=P+Q
```



The volume of S is given by:

```
sage: f(u)=u^2+1
sage: 2*pi * integral(u*f(u),u,0,3)
99/2*pi
```

Note: The volume in this example can be found by washer method:

$$f(u) = u^2 + 1 \Leftrightarrow u = \sqrt{f(u) - 1}$$

$$u = 0 \Rightarrow f(u) = 1, \;\; u = 3 \Rightarrow f(u) = 10$$

where the volume is the sum of rotating the region between $x = 3$ and $x = \sqrt{y-1}$ and the region between $x = 3$ and $x = 0$.

```
sage: var('y')
y
sage: pi*integral((9-(y-1)),y,1,10)+pi*integral(9,y,0,1)
```

```
99/2*pi
```

Those answers agree to each other as they suppose to be.

Example 6.3.4. Sketch the ellipse $\frac{x^2}{a^2}+\frac{y^2}{b^2}=1$ and find the volume of the solid obtained by revolving the region enclosed by the ellipse about the x-axis.

Solution:

```
sage: x,y=var('x,y')
sage: a=1
sage: b=2
sage: h=implicit_plot(x^2/a^2+y^2/b^2==1,(x,-a-1,a+1),(y,-b-1,
    b+1),figsize=3)
```



To plot the corresponding solid of revolution ellipsoid, we first solve the equation $\frac{x^2}{a^2}+\frac{y^2}{b^2}=1$ for y

```
sage: var('a,b')
(a, b)
sage: sol=solve(x^2/a^2+y^2/b^2==1,y)
sage: sol
[
```

```
y == -sqrt(a^2 - x^2)*b/a,
y == sqrt(a^2 - x^2)*b/a
]
```

The positive and negative of y above correspond for the upper haft and lower haft of the ellipse. Let consider the upper haft in plotting and computing the volume of the ellipse. Define:

$$f(x) = \sqrt{b^2 - \frac{b^2x^2}{a^2}} = b\sqrt{1 - \frac{x^2}{a^2}}$$

```
sage: f(x)=sol[1].rhs()
sage: f(x)
sqrt(a^2 - x^2)*b/a
sage: f(x)=f(x).substitute(a=1,b=2)
sage: S=revolution_plot3d(f(x),(x,-1,1), show_curve=True,
   opacity=7, parallel_axis='x' )
```



Since the ellipsoid is defined on the interval $[-a, a]$, its volume S based on the disc method is:

```
sage: pi*integral(f(x)^2,x,-1,1)
```

```
16/3*pi
```

In general, the volume of the ellipsoid for arbitrary positive a and b is:

```
sage: var('a,b,x,y')
(a, b, x, y)
sage: F(x)=sol[1].rhs()
sage: pi*integral(F(x)^2,x,-a,a)
4/3*pi*a*b^2
```

Thus,

$$\mathrm{Vol} = \frac{4}{3}\pi a b^2$$

Notice that if $a = b$, then the ellipsoid becomes a sphere and the volume will be $\mathrm{Vol} = \frac{4}{3}\pi a^3$ where a is the radius of the sphere.

Bibliography

[1] William Stein. *Sage Quick References: Calculus*. Web. September 2015

[2] Gregory V. Barg. *Sage for Undergraduate*.University of Wisconsin|Stout, Menomonie, Wi, 54751

[3] Sage Tutorial v6.9. *Sagemath.org*. Web. September 2015

Appendices

Appendix A

Common Mathematical Operations

Operation	Traditional Notation	Sage Notation
Define a function	$f(x) = x^2$	$f(x) = x \wedge 2$
Evaluate a function	$f(1)$	$f(1)$
Square root	$\sqrt{f(x)}$	$sqrt(f(x))$
Absolution value	$\|f(x)\|$	$abs(f(x))$
Limit	$\lim_{x \to a} f(x)$	$limit(f(x), x = a)$
Derivative	$f'(x)$	$diff(f(x), x)$
Second derivative	$f''(x)$	$diff(f(x), x, 2)$
Indefinite integral	$\int f(x)dx$	$integral(f(x), x)$
Exact definite integral	$\int_a^b f(x)dx$	$integral(f(x), x, a, b)$
Approximate integral	$\int_a^b f(x)dx$	$n(integral(f(x), x, a, b), digits = 2)$
Pi	π	pi
Euler number	e	e
Imaginary number	i	i
Infinity	∞	infinity
Cosine function	$\cos x$	cos(x)
Inverse cosine function	$arccosx$ or $\cos^{-1} x$	arccos(x)
Exponential function	e^x	exp(x) or $e \wedge 2$
Natural logarithm (base e)	$\ln x$	log(x)

Appendix B

Useful Commands for Plotting and Algebra

Description	Sage Command
Plot a function $f(x)$ over interval $[a, b]$	plot(f(x),x,a,b)
Plot contour of $f(x, y)$ on $[a, b]x[c, d]$	contour_plot$(f(x, y), (x, a, b), (y, c, d))$
Plot an ellipse has center at (x_0, y_0) with radii r_1, r_2	ellipse$((x_0, y_0), r_1, r_2)$
Solve equation $f(x) = g(x)$ for x	solve(f(x)==g(x))
Reduce *expression* to most simple	(expression).simplify.full()
Numerical approximation of a quantity	n(expression) *expression*